
1. Позиционные системы счисления

Позиционные системы исчисления – это системы, в которых количественная сторона объектов определяется формулой:



,

где


 - основание системы счисления: 2,8,10,16...;

i - номер позиции;

xi - цифра в i-ой позиции.

Количество цифр в системах счисления соответствует основанию:

	Двоичная:
	2
	→
	0,1

	Восьмеричная:
	8
	→
	0,1,2,3,4,5,6,7

	Десятичная:
	10
	→
	0,1,2,3,4,5,6,7,8,9

	Шестнадцатеричная:
	16
	→
	0,1,2,3,4,5,6,7,8,9, A, B, C,  D, E,  F

	
	
	
	                               10,11,12,13,14,15


Примеры:

	4
	3
	2
	1
	0

	0
	1
	1
	0
	12


→
1*20 + 0*21 + 1*22 + 1*23 = 1310;

	2
	1
	0

	7
	3
	18


     
  →
1*80 + 3*81 + 7*82 = 47310;

	3
	2
	1
	0

	А
	3
	8
	F16


  →
15*160 + 8*161 + 3*162 + 10*163 = 4234410.

Наиболее оптимальной является система с основанием 

 или ближайшая к ней троичная система. В 70-е годы в СССР находилось в эксплуатации ряд ЭВМ, работающих в этой системе. Разрабатывались  экспериментальные варианты ЭВМ на основе десятичной арифметики, как наиболее близкой для расчетной деятельности человека. Однако дальше уникальных образцов работы не пошли, поскольку конструктивная реализация процессов этого типа была сравнительно сложной. В настоящее время компьютерное обеспечение базируется на двоичной арифметике.

2. Переводы чисел из одной системы счисления в другую

Наиболее часто в практике встречается необходимость следующих преобразований:

Будем использовать два варианта перевода чисел:
вариант 1 – использование принципа позиционных систем. Как правило, вариант используется при переходе в десятичную систему.

вариант 2 – последовательное деление переводимого числа на основание степени системы, в которую переводится  число. Деление реализуется по правилам исходной системы. Результат записывается в виде набора остатков.

Примеры:

1. 1310 → ?2

	13
	2
	
	

	12
	6
	2
	

	(1)
	6  
	3
	2

	
	(0)
	2  
	(1)

	
	
	(1)
	


1310 → 11012
3. 47310 → ?16

	473
	16
	

	32
	29
	16

	153
	16 
	(1)

	144
	(13)
	

	(9)
	
	


47310 → 1D916
2. 47310 → ?8

	473
	8
	

	40
	59
	8

	73
	56 
	(7)

	72
	(3)
	

	(1)
	
	


47310 → 7318
4. 7318 → ?10

	731
	12
	

	62
	57
	12

	111
	50 
	(4)

	106
	(7)
	

	(3)
	
	


7318 → 47310

Достаточно просто реализуются переводы в системах с основанием 2n:

2      8     16

Базируются эти переводы на основе информационных единиц:

Триада - 3-х битовый эквивалент восьмеричной цифры (3 разряда):

	0
	-
	000
	
	4
	-
	100

	1
	-
	001
	
	5
	-
	101

	2
	-
	010
	
	6
	-
	110

	3
	-
	011
	
	7
	-
	111


Триады используются при переводах в восьмеричную систему.

Тетрада - 4-х битовый эквивалент шестнадцатеричной цифры (4 разряда):

	0
	-
	0000
	
	A
	-
	1010

	1
	-
	0001
	
	B
	-
	1011

	2
	-
	0010
	
	C
	-
	1100

	…
	
	
	
	D
	-
	1101

	8
	-
	1000
	
	E
	-
	1110

	9
	-
	1001
	
	F
	-
	1111


Переводы осуществляются путем разбиения исходного двоичного числа на триады или тетрады.

Пример:
	3
	6
	6
	7
	7
	5
	5
	36677558

	0
	1
	1
	1
	1
	0
	1
	1
	0
	1
	1
	1
	1
	1
	1
	1
	0
	1
	1
	0
	1
	

	
	F
	6
	F
	E
	D
	F6FED16


Перевод из шестнадцатеричной системы в восьмеричную осуществляется через двоичную форму.

3. Арифметические операции в системах счисления
Все операции (+, –, *, /) реализуются аналогично десятичной системе счисления.

Примеры:

1. В двоичной системе:
 1011          10010          1011 = 8 + 2 + 1= 11

+0111          – 111         * 111 = 7

10010           1011          1011

                            +1011

                           +1011  
                           1001101 = 64+8+4+1 = 77

2. В восьмеричной системе:

  773          1242          773

 +247         + 773         –247
 1242           247          524

3. В шестнадцатеричной системе:

  1АЕ           BD9          ABCD
 +А2В          –1AE         –0FFF
  ВD9           A2B          9BCE

Умножение и деление выполняются с предварительным переходом к десятичной арифметике, поскольку отсутствует практика использования соответствующих таблиц умножения.

Особый случай представляет умножение и деление на 2 или 2n. Операция реализуется путем сдвига двоичного числа влево при умножении на 2 и вправо при делении на 2, например:

0101*21 = 1010 = 5*2 = 10

0101*23 = 101000 = 5*8 = 40

Особенность операции сдвига дает возможность эффективно реализовать умножение и деление на малые числа. Поскольку операция сдвига очень быстродействующая, то целесообразнее использовать эту операцию, чем специальную команду умножения (MUL).

Пример:

205*17 = 3485 → 205*(24 + 20) = 3485

20510 → ?16

_205 |16      CD → 11001101*17 → 1100 1101 0000 *24

 16  |12                        +     1100 1101 *20

 _45                             1101 1001 1101
  32                               D    9    D

  13

 Проверка: D9D = 13*160 + 9*161 + 13*162 = 3485.

4. Логические (битовые) операции

В микропроцессорах используются следующие логические операции:

OR
– сложение (поразрядная операция "ИЛИ"):

0 + 0 = 0,
1 + 0 = 1,
0 + 1 = 1,
1 + 1 = 1;

AND
– умножение (поразрядная операция "И"):

0 * 0 = 0,
1 * 0 = 0,
0 * 1 = 0,
1 * 1 = 1;

NOT
– отрицание (поразрядная операция "НЕ"):

0 ( 1,
1 ( 0;
XOR
– исключающее "ИЛИ", отрицание равнозначности, сложение по модулю 2:


0 x 0 = 0,
1 x 0 = 1,
0 x 1 = 1,
1 x 1 = 0.

Сложение по модулю 2 реализуется как обычное сложение, но без переноса в старшие разряды:

0 + 0 = 0;


0 + 1 = 1;

1 + 0 = 1;


1 + 1 = 0.

К битовым операциям относятся операции поразрядного сдвига влево и вправо – SHL, SHR.

Алгоритмически можно выделить три битовые операции:

1. Выделение. Из заданного кода необходимо выделить какую-нибудь его часть. Операция реализуется командой AND с подбором соответствующей константы. Константа должна содержать “1” только в тех разрядах, которые нужно выделить.

Примеры:

	*
	*
	*
	*
	
	*
	*
	*
	*
	
	*
	*
	*
	*
	
	*
	*
	*
	*


1. В ячейке А имеется код:

А=


	1
	0
	0
	0
	
	0
	0
	0
	0
	
	0
	1
	1
	1
	
	1
	1
	1
	1


Константа:



С=


C=                8
       0

   7
        F

Результат: R = A  and  807F, если A = ABCD, результат будет равен:

1010 1011 1100 1101

1000 0000 0111 1111
1000 0000 0100 1101 = 804D

2. Приформирование. Обратная операция выделению, когда необходимо из двух разных кодов сделать один общий код в определенной структуре разрядной сетки. Реализуется операцией OR (поразрядное сложение). Этой операции могут предшествовать команды выделения и сдвига. Сама операция используется при создании составных информационных слов, когда в разрядной сетке располагаются несколько физических величин.

Пример:

Пусть имеется три физические величины: A, B, C, пределы которых заведомо известны:

А = 0 ( 9
→ 1001
(**** 
– 4 разряда)

В = 0 ( 100
→ 110 0100
(*** **** 
– 7 разрядов)

С = 0 ( 1
→ 1

(* 

– 1 разряд)

     
Для хранения величин А, В, С требуется 3 байта. Рассмотренные физические величины можно записать в слове, например:

R = * * * *    * * * *   * * * *   * * * *
       C         B                                 A

Формирование может быть выполнено операциями

R = A or (B shl 8) or (С shl 15)

3. Сравнение. Имеются два состояния некоторой величины. Как правило, это шкалы, где каждый разряд имеет свой физический смысл. Операция сравнения должна дать информацию об изменении нового состояния по отношению к старому. Операция реализуется командой XOR. 

Пример:

Пусть имеются состояния
OLD  = 1101111100
xor 






NEW = 1100011100






   0001100000 (1 – изменение состояния)

5. Единицы информации. Формат отрицательных чисел

Различают следующие единицы информации:

бит – (1 разряд); триада (3 разряда); тетрада (4 разряда);

байт – (8 разрядов, 2 тетрады) – минимальный адресуемый эл-т памяти;

слово – (16 разрядов, 4 тетрады, 2 байта);

двойное слово – (32 разряда, 8 тетрад, 4 байта).

строка –набор байт переменной длины (до 255 байт);

шкала – последовательность  бит     или     более    крупных,   но одинаковых   структур,   в    которых    каждый   разряд   играет    самостоятельную    роль     (например,   шкала состояний   коммутационных   аппаратов:  «1»-вкл. «0»-откл.).

составное информационное слово – в байте, слове или двойном слове    компонуются поразрядно различные физические величины.
Адресуемым информационным элементом является байт. Процессорная память  организуется в виде набора байт, каждый из которых имеет свой номер – физический адрес.

Примеры записи чисел в информационных единицах:

Пример:

A
( в байт
( 0А

F1E
( в байт
( 1E

F1E
( в слово
( 0F1E 

А
( в слово
( 000А

8A9BCD ( в двойное слово
( 008A9BCD

Представление числовой информации возможно в двух вариантах:

кодовый – числом занята вся разрядная сетка информационной единицы (числа без знака);

числовой – в разрядной сетке должен появиться определенный признак, говорящий о знаке числа (числа со знаком).

Максимальные числа, размещаемые в разрядной сетке, определяются по схеме:


    * * * * * * * * *  max = 29 – 1 = 511.

байт
(
                  1111 1111 ( 28  – 1 = 255;

слово
(
1111 1111 1111 1111 ( 216 – 1 =  65535.

Отрицательные числа записываются в "обратном коде". Возможны два метода получения обратного кода:

– число =

 + 1;

– число = «очень большое число» – число
«Очень большое число» значительно больше любых разрядных сеток информационных единиц.

Примеры:
Метод1 


Метод2


Контроль
5В =
0101 1011


1 … 0 0


   5 B

инверсия 
1010 0100


     ​– 5 B


+ A 5
                                 +1


        A 5


 1 0 0 (байт = 0)

–5В =
1010 0101 = A5

Переход от "обратного кода" к положительному числу выполняется аналогично. Отрицательное число всегда имеет в старшем разряде 1. За счет записи отрицательных чисел в "обратном коде" в архитектуре микропроцессора операция вычитания реализуется операциями инверсии и сложения: 

A – B = A + (–B), например 

А = 55 = 


0011 0111

В = –72 = –0100 1000 =
1011 1000
А + В = 55 + (–72) = 
1110 1111 = –0001 0001 = –17
Максимумы чисел со знаком:

Max байт (+) = 0111 1111 = 127, Max слово (+) = 0111 1111 1111 1111 = 32767,
Max байт (–) = 1000 0000 = 128, Max слово (–) = 1000 0000 0000 0000 = 32768.
Абсолютный максимум отрицательного числа всегда на 1 больше положительного. 

6. Архитектура МП Intel 8086 (8088)

В архитектуре МП имеется 4 группы регистров:

1. Регистры данных


	AX
	AH
	AL
	Аккумулятор

	BX
	BH
	BL
	

	CX
	CH
	CL
	счетчик (организация цикла)

	DX
	DH
	DL
	




2. Регистры указателей и индексов

	SP
	указатель стека

	BP
	указатель базы

	SI
	индексные регистры для работы с массивами, например, POLE[DI]

	DI
	


Стек – специальная организация памяти, запись в которую не уничтожает предыдущее значение, а погружает его на единицу вниз. При выборке из стека последняя запись уничтожается, а предыдущая всплывает.

3. Управляющие регистры

IP –
указатель команд. Используется при выборке команд для выполнения конкретной операции.

RF –
регистр флагов. Содержит шкалу, где каждый разряд имеет специфичное значение. Можно выделить следующие основные флаги:

· флаг нуля;

· флаг знака;

· флаг переполнения и др.

4. Регистры сегментов

В процессорах 8086 шина данных имеет 16 дорожек, а шина адреса – 20 дорожек, поэтому для формирования адреса используются два регистра, называемых сегмент и смещение. Полный адрес вычисляется по выражению:

Адрес = сегмент*24 + смещение, 
например, 

сегмент      = A01F,  А01F0

смещение     = 00FE,  +00FE
полный адрес =        A02EE
максимальный адрес = FFFFF = 1048575 = 1Мбайт.

в качестве сегмента используются регистры:

CS – сегмент команд; 
DS – сегмент данных; 
SS – сегмент стека; 
ES – дополнительный сегмент.


Для разных сегментных регистров используются разные регистры смещения, например:


CS : IP – адрес сегмента кода;


SS : SP – адрес сегмента стека;


DS : BX, DI, SI – адрес сегмента данных.


Смещения имеют адресацию 0, 1, 2, 3, … относительно начала сегмента.
7. Система команд МП

Можно выделить семь групп команд:

1. Пересылка
2. Арифметика
3. Логика
4. Передача управления
5. Обработка строк
6. Обработка прерываний
7. Управление процессором
Любая команда имеет структуру, состоящую из двух частей: 

	КОП
	КАД


КОП – код операции, символическое обозначение действия над операндами. 

КАД – код адреса (адресная часть), которая указывает, над чем должна быть выполнена операция. В адресной части можно выделить две компоненты:
Р – приемник, I – источник.

Система команд МП имеет двухадресную или одноадресную структуру. Формальная запись адресации P и  I имеет различные варианты. Наиболее простой является регистровая адресация, когда в качестве приемника и источника записываются конкретные регистры. Возможны и другие варианты адресации.

Команды пересылки:
MOV

P, I
(I → P)
– пересылка из I в P;
XCHG
P, I
(I ↔ P)
– перемена мест;

PUSH
I
(I → стек)
– запись источника в стек;

POP

P
(стек → P)
– выборка из стека в приемник;

IN

DX, AX

– из порта (DX) в аккумулятор (AX);

OUT

DX, AX

– вывод из аккумулятора (AX) в порт (DX).

Пример:

MOV DI, AX – пересылка из регистра AX в DI;

MOV CL, -30 – пересылка константы в младший байт регистра CX;

MOV AX, TABL – информация из памяти, определенной под идентификатором TABL переписывается в аккумулятор (из памяти берется 2 байта, т.к. в АХ пересылается слово);

MOV CH, TABL – пересылка одного байта;

PUSH DX – содержимое регистра DX погружается в стек (2 байта);

POP AX – выборка из стека 1 слова в регистр AX;

MOV POLE, TABL – недопустимая адресация (память ( память).

Арифметические операции:
ADD
P, I
(P+I → P) – сложение. При   реализации   операции возможно    переполнение, при этом устанавливается флаг переноса или переполнения (CF).

ADС
P, I
(P+I+CF → P) – сложение с учетом флага переноса или переполнения.

Операция позволяет реализовывать сложение двойных слов и более крупных числовых организаций по следующей схеме:


INC
P

(P + 1 
→ P)

SUB
P, I

(P – I 

→ P)

SBB
P, I

(P – I – CF 
→ P)

DEC
P

(P – 1 
→ P)

NEG
P

(-P → P) – инверсия чисел

CMP
P, I

(P – I) – операция вырабатывает только флаги.

MUL
I – операция умножения. Один операнд всегда находится в аккумуляторе (регистр AX). Приемником всегда служит регистры AX и DX.    Операция выполняется в модификациях:

Умножение байт:
AL

I → AX;
Умножение слов:
AX

I → DX, AX;
DIV
I – операция деления.

Деление байт:
AX / I → AL частное, AH остаток;
Деление слов:
DX, AX / I → AX частное, DX остаток;
IMUL, IDIV – умножение и деление чисел со знаком.
Логические операции:
AND
P, I

(P 

I → P)

OR
P, I

(P 

I → P)

XOR
P, I

(P 

I → P)

NOT
P

(
[image: image1.wmf]P

 → P) – инвертирование

TEST
P, I

(P 

 I) – вырабатываются только флаги.

К логическим операциям также отнесем операции поразрядных сдвигов:
SHL, SHR – сдвиг "влево", сдвиг "вправо". Выпадающий разряд попадает в флаг переноса, освободившийся разряд заполняется нулем.
ROL, ROR   –   циклические операции сдвига.


RCL, RCR
Количество требуемых сдвигов записывается в регистре СL. Если необходим сдвиг на один разряд, то в адресной части записываем 1, например, SHL BX, 1 (сдвиг "влево" на один разряд), SHR AX, CL (сдвиг "вправо" на CL разрядов).
Примеры арифметических и логических операций:

1.  ADD AX, DX

AX=BBBB



  BBBB
DX=FF13



+FF13
1BACE

Результат: AX = BACЕ, флаг переноса CF = 1

2.  XOR AX, DX

AX = F0F0

DX = 0F0F 

Результат: AX = FFFF

3.  IMUL DI

AX = 00A3



A3 = 3*16

+ 10*16 = 163 

DI  = FFF2

FFF2 = число отрицательное, переведем его в положительное:

10000000           163


2282 |16
    FFF2         * -14         -16   |142|16
    000E           652


 68  -128|8
                  163 


-64    14=E

                   –2282            42

                                   -32
                                    10=A

FFF2 = -14

2282  = 8EA

Отметим, что результат отрицательный и записывается в два регистра AX и DX (двойное слово). Переведем в отрицательное число: 
1 0 0 0 0 ....0 0 0 0 0 0

   8 Е А

.... F F F F F F F 7 1 6

DX=FFFF

AX=F716
Операции передачи управления:
1. Безусловные передачи

JMP 
MET – передача управления по метке MET;

CALL
PROG – безусловная передача управления с возвратом, используется при обращении к процедурам;

RET – возврат из процедуры по адресу возврата.

При реализации операции CALL адрес возврата автоматически записывается в стек. Этот адрес может быть словом (смещением) или двойным словом (сегмент кода + смещение). По размеру адреса возврата различают ближний (NEAR) и дальний (FAR) тип вызова процедур. В качестве смещения используется счетчик команд IP.
При операции RET из стека автоматически выбирается адрес возврата в сегмент кода и счетчик команд. Связь между процедурами через стек позволяет реализовать произвольную глубину связи различных процедур.



При использовании многоуровневой процедурной программы необходимо внимательно работать со стеком (операции POP, PUSH).
2. Условные передачи


Все условные передачи работают в зависимости от регистра флагов RF:
	15
	14
	13
	12
	11
	10
	9
	8
	7
	6
	5
	4
	3
	2
	1
	0

	
	
	
	
	
	
	
	
	SF
	ZF
	
	
	
	PF    
	
	CF


CF – флаг переноса, 1 – если возник перенос или переполнение;
ZF – флаг нуля, 1 – если результат операции равен 0;
SF – флаг знака, 1 – если результат отрицательный;
PF – флаг четности, 1 – если количество 1 в младших 8 битах четное.
В регистре флагов RF также есть другие биты флагов. Перечисленные флаги устанавливаются при выполнении арифметических или логических команд. Некоторые команды флагов не вырабатывают, например, команда MОV. Имеются специальные команды установки и сброса флагов, например:

STC
(1 → CF);
СLC
(0 → CF);
СMC
(инверсия CF).
Условная передача управления имеет структуру:

J***

JC

MET – переход по метке MET, если CF=1;
JNC

MET – переход, если CF=0;
JZ

MET – переход, если ZF=1;
JNZ

MET – переход, если ZF=0;
JS

MET – переход, если SF=1;
JNS

MET – переход, если SF=0.
Для облегчения программирования есть группа операций, работающих по принципу отношений, которым предшествует операция сравнения CMP:
JG

MET – передача управления, если больше;
JGE

MET – передача управления, если больше или равно;
JL

MET – передача управления, если меньше;
JLE

MET – передача управления, если меньше или равно;
JE

MET – передача управления, если равно;
JNE

MET – передача управления, если не равно.

Пример:

ADD
AL, BL

JC
MET1
– передача по MET1, если было переполнение.

CMP
AL, BL

JE
MET 2
– переход на MET2, если два операнда равны (AL = BL).

3. Циклические передачи

Для организации цикла используется команда LOOP MET. Начальное количество циклов записывается в регистр CX. Команда LOOP реализует вычитание счетчика CX на 1 и проверку на 0. Если не 0, то выполняется переход по MET, в противном случае выполняется переход к следующей команде.
Пример:

MOV
CX, 100

MET:

..........

..........

..........

LOOP
MET



..........
Возможны модификации команды LOOP.

8. Структура программы на языке Assembler

TITLE
.....



;наименование программы
EXTRN
V1:BYTE, V2:WORD, V3:DWORD, VPROC:FAR
PUBLIC
P1, P2, P3, PR2
STEK

SEGMENT
PARA
STACK 
;сегмент стека


DB
64 DUP(?)

STEK

ENDS
DSEG
SEGMENT
PARA

;сегмент данных

P1
DB
123, 1011b, 7Ah, 'строка'


Р2
DW
1FFFh, 0FFA1h

P3
DD
10 DUP(0)

DSEG
ENDS
CSEG
SEGMENT PARA


;сегмент кода

ASSUME
CS:CSEG, DS:DSEG, SS:STEK

start:
MOV
AX, DSEG


;загрузка сегмента данных

MOV
DS, AX


.......................


PR1
PROC
NEAR
;процедура ближнего вызова (NEAR)


.........



RET



;завершение процедуры

PR1
ENDP


.......................


CALL PR1



;вызов процедуры PR1

.......................


PR2
PROC
FAR

;процедура дальнего типа вызова (FAR)


.........



RET


PR1
ENDP


;завершение процедуры

.......................


MOV
AX, 4C00h


;завершение программы

INT
21h

CSEG
ENDS
END
start
Комментарии:
1. Наименование программы указывается после оператора TITLE. Тексты примечаний пишутся после символа ";".
2. EXTRN – оператор указания внешних имен переменных и процедур. Для переменных указывается размерность: BYTE, WORD, DWORD, для процедур – тип вызова: FAR, NEAR (для внешних процедур, как правило, FAR).

3. PUBLIС – оператор указания внутренних имен переменных и процедур для внешнего использования (размерность не указывается).

4. В программе выделяется три сегмента: стека, данных и кода. Имена сегментов (STEK, DSEG, CSEG) произвольны. Каждый сегмент начинается оператором SEGMENT и заканчивается оператором ENDS. Для сегмента стека нужно указывать служебное слово STACK.
5. Внутри сегмента стека должен быть определен его объем. В сегменте данных должны быть определены все внутренние переменные, в том числе константы. Для определения типов переменных в сегменте данных, используются операторы:

DB
– байт;

DW
– слово;

DD
– двойное слово,
для выделения заданного объема памяти используется оператор DUP, например

DB 64 DUP(?) – выделено память размером 64 байта, содержимое памяти не определено;

DD 10 DUP(0) – выделено память размером 10 двойных слов (40 байт), содержимое памяти заполнено нулями.
запись констант имеет вид: 
123
– десятичное число;

1011b
– двоичное число (биты);

7Аh
– шестнадцатеричное число;

0FFA1h – если число начинается с буквы, то впереди должен быть 0;

’строка’ – текстовая константа;
6. Оператор ASSUME реализует привязку наименований сегментов к сегментным регистрам, например:
ASSUME CS:CSEG, DS:DSEG, SS:STEK.

При загрузке программы CS содержит адрес сегмента кода, SS – адрес сегмента стека (за счет оператора STACK), а регистры DS и ES – адрес начала программы (так называемый префиксный сегмент PSP), поэтому в начале текста программы должна быть стандартная часть загрузки регистра DS адресом сегмента данных, например:
MOV
AX, DSEG
MOV
DS, AX

7. В сегменте кода могут быть реализованы процедуры, например, PR1, PR2. Процедуры используемые в этом сегменте кода (PR1) могут иметь ближний тип вызова (NEAR), а процедуры передаваемые во внешнее пользование должны иметь дальний тип вызова (FAR). Процедура должна заканчиваться оператором RET.
8. Программа завершается вызовом прерывания INT 21h, которому передается номер функции AH = 4Ch (завершение программы) и код выхода AL = 00h (завершение без ошибок).
9. После оператора END нужно указать точку входа в программу –название метки или процедуры в сегменте кода.
9. Пример 1. Построение гистограммы нагрузок.

Необходимо составить программу статистической обработки данных некоторой нагрузки, построить гистограмму заданной выборки. Нагрузки размещены в массиве TABL и имеют технологические ограничения, например, могут изменяться от 0 до 100 кВт. Если измерения выходят за технологические пределы, они считаются аномальными и игнорируются. Максимальный объем массива 1000 байт. Заполняет массив TABL внешняя процедура, связанная с системой опроса датчиков. Реальный объем измерений будем определять через внешнюю переменную N. Гистограмма размещается в массиве H1ST длиной 101слово (точность измерения - 1 кВт).

Следует обратить внимание на параметрическую адресацию массива HIST.


Если Р=0

Если Р=1

Если Р=2

Если Р=3

Если Р=100

Адресация к гистограмме имеет вид:
НIST[2

P].

Алгоритм

0 → HIST

CХ = N
<TABL[CХ] >100>

<TABL[CХ] < 0>

P = TABL [CХ]

P = P*2
HIST[P] = HIST[P] +1

<CX – 1, CX = 0>

Программная реализация
TITLE
PRIM 1

EXTRN
N: word
public
tabl, hist, gist
DSEG
SEGMENT

tabl
DB
1000 DUР(0)


Hist

DW
101 DUP(0)

DSEG
ENDS

CSEG
SEGMENT

ASSUME
DS: DSEG, CS: CSEG

GIST

PROC
FAR



MOV

AX, DSEG
; загрузка сегмента данных


MOV

DS, AX
MOV

CX, 101
; обнуление массива HIST
MET1:
MOV

DI, CX

SHL

DI, 1

DEC

DI

DEC

DI

MOV

HIST[DI], 0

LOOP
MET1

MOV

CX, N
; выборка значения мощности в AL
MET2:
XOR

AX, AX

MOV

DI, CX

DEC

DI

MOV

AL, TABL[DI]

CMР

AL, 100
; анализ ограничений
JG

MET3

CMP

AL, 0

JL

MET3

MOV

DI, AX
; суммирование гистограммы
SHL

DI, 1

INC

HIST[DI]

MET3:
LOOP
MET2
RET

GIST

ENDP

CSEG
ENDS

END
Комментарии:

1. Величина N – количество измерений является внешней переменной.
2. Для внешнего использования объявлены массивы TABL и HIST и процедура GIST.
3. Несмотря на первичное зануление массива HIST при его создании, в программной реализации есть собственное зануление, т.к. возможно повторное обращение к процедуре;

4. Операция обнуления аккумулятора XOR AX, AX необходима для корректности пересылки MOV DI, AX, поскольку возможна ситуация, когда AH ≠ 0;

5. В примере используется параметрическая адресация, когда индекс (регистр DI) имеет физическую размерность (кВт).

Реализация управляющей программы к примеру 1

TITLE
MAIN_PRIM1

extrn
tabl:bYte, hist:word, gist:far
Public
N

STEK

SEGMENT
STACK
DB
512 DUР(0)

STEK

ENDS

DSEG
SEGMENT

N
DW
0

DSEG
ENDS

CSEG
SEGMENT

ASSUME
DS: DSEG, CS: CSEG, SS: STEK
start:
MOV  AX, DSEG

MOV  DS, AX

………………
; Загрузка массива TABL
CALL  GIST

………………
; Выборка HIST

MOV
AX, 4C00h

INT
21h
CSEG
ENDS
END
start
Комментарии:
1. Необходимо обратить внимание на симметричность операторов EXTRN и PUBLIC в управляющей и подчиненной процедурах:
EXTRN


EXTRN

PUBLIC


PUBLIC

2. Для определения сегмента стека использовано служебное слово STACK.

3. Метка start использована для определения точки входа в программу.
4. Загрузка TABL может быть выполнена различными методами (работа с портами, генератор случайных чисел, обозримый массив констант).

5. Выборка HIST должна быть связана с выдачей информации на устройства ввода-вывода.
10. Пример 2. Программирование табличных функций.

Необходимо разработать программу расчета sin α, опираясь на табличные функции. В принципе, можно использовать стандартное разложение

[image: image2.wmf]sin

α = α – α3/3! + α5/5! – α7/7! + ..., где α - аргумент в радианах.

Однако, при отсутствии операций с плавающей точкой реализовать эту функцию сложно. Кроме того, время вычисления будет относительно большим. Если требуется максимальное быстродействие и если есть возможность ограничиться по точности (например, 1 градус) можно реализовать задачу с табличной функцией. Введем массив значений sin α для α = 0 ... 90°.
TABSIN
DW
0,
175,
349,
523,
689,
....
10000

α
0,
1°,
2°,
3°,
4°,
....
90°

Отметим, что величины sin α записаны в масштабе 10-4. Хранить значения массива для 360° нет необходимости (значения повторяются). Для знака введем переменную ZNAK (0 – знак "+", 1 – знак "–").
Алгоритм




Результатом является составное число, где старший разряд показывает знак синуса. Остальную разрядную сетку занимает значение синуса. Решение является корректным, поскольку 10000 → 271016 и знаковый разряд никогда не заполняется.

Условимся, что входная величина α при обращении к процедуре задается в аккумуляторе AX. Переменную ZNAK будем хранить в регистре CX. Результат работы процедуры sin α будем передавать через регистр ВХ. Необходимо предусмотреть запоминание и восстановление требуемых регистров. Структурно sin α является составным информационным (а не числом со знаком).

Программная реализация

TITLE
PRIM2

PUBLIC
SIN

DSEG
SEGMENT

TABSIN
DW ...

DSEG
ENDS

CSEG
SEGMENT

ASSUME
DS:DSEG, CS:CSEG

SIN
PROC
FAR

MOV

BX, DSEG
; загрузка сегмента данных
MOV

DS, BX
PUSH
AX

; сохранение регистров в стеке
PUSH
CX

PUSH
DI

XOR

CX, CX
; обнуление знака
CMP

AX, 181

JL

MET1

MOV

CX, 8000h
; ZNAK = 1 (знак "–")
SUB

AX, 180

MET1:
CMP

AX, 91

JL

MET2

NEG

AX

ADD

AX, 180
MET2:
MOV

DI, AX

SHL

DI, 1

; DI*2
MOV

BX, TABSIN[DI]

OR

BX, CX
POP

DI

; восстановление регистров
POP

CX

POP

AX

RET

SIN
ENDP

CSEG
ENDS

END
Комментарии:

1. В процедуре отсутствует оператор EXTRN, т.к. нет требуемых внешних имен.
2. Константа 8000h – единица в старшем разряде слова.
3. В процедуре реализовано заполнение и восстановление состояния регистров AX, CX, DI.
11. Пример 3. Обработка битовых шкал телесигналов.


Необходимо составить программу обработки состояний коммутационных аппаратов некоторого фрагмента электрической сети. Пусть по каналам телеметрии с определенной периодичностью осуществляется опрос телесигналов с коммутационных устройств подстанций. Сигнал ТS = 1 соответствует включенному состоянию, ТS = 0 – отключенному. Состояние каждого коммутационного аппарата занимает один разряд. Совокупность этих разрядов организовывает шкалу, которая заносится в массив NEW.


                             16                       8 7 6 5      4 3 2 1

NEW + 0      ****   ****   ****   ****

NEW + 2      ****   ****   ****   ****
NEW + 4      ****   ****   ****   ****
Номер коммутационного аппарата и номер ячейки связаны формулой: 
N = i*16/2 + R,
где
i – номер ячейки массива 0, 2, 4, 6 ...;

N – номер коммутационного аппарата ТS;

R – смещение по слову в диапазоне 1, 2, ... 16, например, для i = 4 и R = 5
N=4*16/2 + 5 = 37.
Аналогично NEW существует массив OLD, который хранит предыдущее значение шкалы аппаратов. В результате анализа необходимо сформировать массив VAR, содержащий перечень только изменившихся коммутационных аппаратов. Информация о каждом таком аппарате будет составным информационным словом.
	
VAR + 0
	TS1
	N1

	
VAR + 2
	TS2
	N2

	
VAR + 4
	TS3
	N3


где
N – номер коммутационного аппарата, изменившего состояние;

TS – собственно новое состояние этого измерения.

После обработки массив NEW переписывается в OLD и хранится до следующего периода опроса, т.е. имеем простейшую задачу реального времени.

Алгоритм анализа

Введем индексы:

i – индекс движения по массиву NEW, OLD = 0, 2, 4, ...;

j – индекс движения по массиву NEW, OLD с шагом 16: j = 0, 16, 32, 48, ... L, где L – суммарное количество коммутационных аппаратов. При этом связь с номером текущего аппарата принимает вид: N = j + R, где R – индекс движения по слову (16, 15, 14, ... 1).

k – индекс по массиву VAR (0, 2, 4...) или  количество изменившихся ТS.

Механизм сравнения и анализа заключается в следующем:

1. Операцией XOR сравниваем старое и новое значение шкалы. В результате имеем новую шкалу, в которой разряд «1» свидетельствует об изменении TS.
2. Анализ конкретного изменения будем выполнять по старшему разряду сравнения, используя операцию сдвига: например,

NEW
1 0 1 1 ...

OLD
0 0 0 1 ...

XOR
1 0 1 0 ...
→
TS16 изменило состояние

SHL
0 1 0 ...
→
TS15 не изменило состояние

SHL
1 0 ...

→
TS14 изменило состояние

SHL
0 ...

→
TS13 не изменило состояние

Параллельно с анализом будем сдвигать шкалу NEW для выборки нового значения TS:
  NEW

1 0 1 1
→
TS16=1

SHL
0 1 1

→
…

SHL
1 1

→
TS14=1

SHL
1

→
...

Алгоритм
i = 0, j = 0, k = 0
Start:

<NEW[i] = OLD[i]>

R = 16
T1 = NEW[i]
T2 = T1 xor OLD[i]
MET3:
< T2 ( 0 >

VAR[k] = (T1 and 8000h) or (j +R)
k+2

MET4:
shl T1

shl T2
MET2:
<R – 1, R = 0>

NEW[i] ( OLD[i]

MET1:
i+2
< j+16, j < L >

При реализации условимся, что массивы NEW, OLD, VAR, а также  длина L являются внешними именами для данного модуля. Результирующая длина массива VAR (переменная k) будет внутренней переменной, которую нужно передать во вне. Индексы будем хранить в регистрах: i ( DI, k ( SI, R ( CX.
В составе собственных переменных должны быть величины T1, T2, k, N, j.
Программная реализация
TITLE
PRIM3

EXTRN
NEW:WORD, OLD:WORD, VAR:WORD, L:WORD

PUBLIC
k, ComApp
DSEG
SEGMENT

T1
DW
0

T2
DW
0

N
DW
0

k
DW
0

j
DW
0

DSEG
ENDS

СSEG
SEGMENT

ASSUME
DS:DSEG, CS:CSEG

ComApp
PROC
FAR

MOV

AX, DSEG
; загрузка сегмента данных
MOV

DS, AX

XOR

DI, DI
; обнуление индексов
XOR

SI, SI

MOV

j, SI

Start:

MOV

AX, NEW[DI]; если NEW = OLD ( MET1
XOR

AX, OLD[DI]

JZ

MET1

MOV

CX, 16
; выборка телеизмерений
MOV

AX, NEW[DI]

MOV

T1, AX

XOR

AX, OLD[DI]; анализ изменения конкретного TS
MOV

T2, AX

MET3:
CMP

AX, 0

JGE

MET2

MOV

AX, j

; расчет номера N
ADD

AX, CX

MOV

N, AX

MOV

AX, T1
; загрузка массива VAR
AND

AX, 8000h
OR

AX, N

MOV

VAR[SI], AX

INC

SI

INC

SI

MET4:
SHL

T1, 1

; переход к следующему TS
SHL

T2, 1

MET2:
LOOP
MET3

MOV

AX, NEW[DI]
; пересылка NEW ( OLD
MOV

OLD[DI], AX

MET1:
INC

DI

INC

DI

ADD

j, 16

MOV

AX, j
CMP

AX, L

; проверка конца шкалы TS
JL

Start
SHR

SI, 1

MOV

k, SI

RET

ComApp
ENDP

CSEG
ENDS

END

Комментарии:
При программировании на языке Assembler оба операнда в двухадресной процедуре не могут быть типа «память». Поэтому во многих местах реализации операции «память - память» реализуются с помощью промежуточных пересылок через аккумулятор.

12. Пример 4. Обработка данных электропотребления.

Необходимо запрограммировать учет потребления электроэнергии. В системах учета и контроля электроэнергии имеется МП-контроллер, который собирает информацию со счетчиков электрической энергии, накапливает ее, предоставляет возможности просмотра на экране и передает данные в центр управления энергоснабжающей компании. Будем считать, что имеются счетчики со встроенными импульсными датчиками.

В системе возможны два варианта опроса датчиков:
1. Асинхронный – как только поступает сигнал из любого счетчика (выставляется «1» в соответствующий разряд порта), реализуется прерывание текущей задачи, управление передается программе обработки прерывания, которая анализирует, какой счетчик вызвал прерывание, и подсуммирует масштабную единицу данных об электропотреблении.

Как правило, контроллер кроме обработки показаний счетчика выполняет целый ряд функциональных задач. Например, формирует суточный график потребления электроэнергии предприятий, составляет баланс электроэнергии, определяет превышение лимита потребления и т.д. Работа такой системы в асинхронном режиме требует сравнительно сложной операционной системы, более сложных конструктивных решений МП-системы и удорожание контроля.
2. Синхронный – контроллер с заведомо известным интервалом времени опрашивает порт импульсных сигналов. Выполняется обработка всей шкалы и прибавляются «1» в требуемые элементы массива данных.


∆t – интервал опроса, выбирается опытным путем, определяется темпом электропотребления, когда один и тот же счетчик гарантированно не успевает выставить две единицы.

τ –
время обработки шкалы счетчиков, должно быть минимальным, чтобы дать возможность минимизации ∆t и оставить достаточное время для фоновых задач.

 (∆t – τ) = время фоновых задач.

Реализация требует от программистов максимальных усилий.

Естественное решение
Информация об электропотреблении пусть хранится в массиве METER, в котором для каждого счетчика отводится одно слово. Прием из портов при опросе записывается в два регистра DX и AX, т.е. контроллер имеет возможность работать с 32 счетчиками:
	№ сч
	
	METER
	
	Порт

	1
	   0
	320
	+
	0
	DX

	2
	2
	55
	+
	1
	

	3
	4
	67
	+
	1
	

	...
	...
	…
	
	…
	

	16
	30
	156
	+
	0
	

	17
	32
	1765
	+
	0
	AX

	18
	34
	247
	+
	1
	

	19
	36
	83
	+
	0
	

	...
	...
	…
	
	…
	

	32
	62
	97
	+
	1
	


Для переадресации будем использовать два индексных регистра:

DI  = 32, 30, 28 ...2 (шаг 2);

BX = 32, 0, -32 (шаг 32).

Адресация к массиву METER имеет вид: METER[DI+BX–2]. Здесь суммарное смещение по массиву METER будет равно: 62, 60, 58, ...0, при этом заполнение будет начинаться с конца.
Алгоритм

BX = 32
DI = 32
shl AX

<CF ≠ 1>

METER [DI+BX-2] + 1
<DI – 2, DI = 0 >

AX = DX
BX – 32
<BX ( 0>
Программная реализация

…………………
MOV

BX, 32

MET1:
MOV

DI, 32

MET2:
SHL

AX, 1

JNC

MET3

INC

METER [DI+BX–2]

MET3:
DEC

DI

DEC

DI

JNZ

MET2

MOV

AX, DX

SUB

BX, 32

JGE

MET1

Решение на основе блочной арифметики

В курсе микропроцессорной техники отмечалось, что арифметическая операция «сложение» может быть реализована логическими операциями XOR и AND: операция XOR дает результат, а операция AND со сдвигом влево на один разряд дает значение переноса, затем операции XOR и AND нужно повторить с результатом и переносом до тех пор пока перенос станет равным нулю.
Пример:
F + A = 15 + 10 = 25
	
	
	XOR
	AND
	

	F
	1111
	
	
	


	A
	1010
	
	
	

	
	00101
	0101
	1010
	есть перенос

	
	10100
	
	10100
	сдвиг (

	
	10001
	10001
	00100
	есть перенос

	
	01000
	
	01000
	сдвиг (

	
	
	11001
	00000
	нет переноса

	
	
	
	
	

	25 =
	11001
	
	
	


Примем для решения задачи другую организацию памяти. Развернем массив METЕR на 90(. При этом в первых четырех байтах будут расположены все младшие разряды для всех счетчиков. Во вторых четырех байтах – все вторые разряды и т.д. Такая организация памяти дает возможность исключить цикл по счетчикам и одним сложением реализовать операцию суммирования по всем счетчикам.
	
	
	AX
	DX

	
	
	
	
	
	
	
	
	
	
	
	

	METER
	   0
	
	
	
	
	
	
	
	
	
	0
	– младший бит

	
	4
	
	
	
	
	
	
	
	
	
	1
	

	
	8
	
	
	
	
	
	
	
	
	
	2
	– значение счетчика

	
	12
	
	
	
	
	
	
	
	
	
	3
	

	
	…
	
	
	
	
	
	
	
	
	
	…
	

	
	60
	
	
	
	
	
	
	
	
	
	15
	– старший бит

	№ сч
	
	32
	31
	30
	…
	17
	16
	15
	14
	…
	1


Будем операцию сложения выполнять на основе логики с контролем переноса.

Пример1:


Значение счетчика равно 3, при опросе порта имеем 1 для этого счетчика, в результате значение счетчика = 3 + 1 = 4. 
       порт
METER  AX,DX   AND    XOR  METER
...1     1      1      0   ...0
...1     1      1      0   ...0
...0     1      0      1   ...1
....                       ....
   3                          4
Пример2:

Пусть имеется три счетчика, содержащих соответственно три значения: 1,3,7. В результате опроса получено изменение 1 и 3 счетчиков, т.е. имеем входной код = 1 0 1:
           порт
METER      AX,DX    AND     XOR    METER
...1 1 1   1 0 1   1 0 1   0 1 0   ...0 1 0
...0 1 1   1 0 1   0 0 1   1 1 0   ...1 1 0
...0 0 1   0 0 1   0 0 1   0 0 0   ...0 0 0
...0 0 0   0 0 1   0 0 0   0 0 1   ...0 0 1
........                           ........
   1 3 7                              2 3 8
проверка:   1 3 7
           +1 0 1
            2 3 8
Таким образом, новое значение METЕR равно 2, 3, 8
Программная реализация

По массиву METЕR возможно движение в случае переноса с шагом 4. Для индексации выберем регистр DI. Исходная шкала лежит в AX, DX.
XOR

DI, DI
MET:

MOV

R1, METЕR[DI]


MOV

R2, METЕR[DI+2]

XOR

METЕR[DI], AX

XOR

METЕR[DI+2], DX

ADD

DI, 4

AND

AX, R1

AND

DX, R2

JNZ

MET

OR

AX, 0
JNZ

MET
Комментарии:


1. R1, R2 – некоторые регистры процессора;


2. В R1 и R2 запоминается состояние двойных слов массива METER, поскольку нижележащей операцией XOR это состояние затирается.


3. Выход из цикла осуществляется по контролю за нулевым переносом и реализуется двумя командами JNZ. Каждая из этих команд соответственно анализирует на ноль содержимое регистров DX и AX.
4. Данная программа работает ориентировочно на 30% быстрее, чем естественная реализация.
13. Пример 5. Умножение двойных слов.
Система команд процессора не обеспечивает возможность умножения двойных слов (MUL умножает только слова). Поэтому задача реализуется программным путем.


 2cл.

      2сл.

  4сл.

Умножение необходимо выполнять по частям и результаты произведений складывать в соответствии с разрядной сеткой. Исходные множители будем размещать в памяти под следующими наименованиями:

DW1L – младшая часть Mн.1 (слово),
DW1H – старшая часть Mн.1,
DW2L – младшая часть Mн.2,
DW2H – старшая часть Mн.2.
(DW1L+DW1H*216)*(DW2L+DW2H*216) = (DW1L*DW2L)+

+(DW1L*DW2H)*216+(DW1H*DW2L)*216+ (DW1H*DW2H)*232
Если обозначить 
N1 = DW1L*DW2L (двойное слово), N1H, N1L – старшая и младшая часть N1,
N2 = DW1L*DW2H, 

N3 = DW1H*DW2L, 
N4 = DW1H*DW2H, то суммирование этих составляющих будет выполняться по схеме:



+0

+2

+4

+6


Все переменные – DW1L, DW1H, DW2L, DW2H, REZ, будем размещать в сегменте данных реализуемой процедуры. Передача множителей выполняется из вне, и результат также должен быть использован во внешней процедуре, поэтому эти переменные должны быть описаны в разделе PUBLIC.
Результат сложения записывается в память в поле REZ, содержащее 4 слова, REZ+0 – старшее слово, REZ+6 – младшее слово. При сложении необходимо учитывать возможности сквозного переноса разрядов в поле REZ+2 и REZ+0.

Программная реализация

TITLE
PRIM5

PUBLIC
DW1L, DW1H, DW2L, DW2H, REZ, MULD
DSEG
SEGMENT

DW1L
DW
DW1H
DW
DW2L
DW
DW2H
DW
REZ

DW
4  DUP(0)

DSEG
ends
CSEG
SEGMENT

ASSUME
DS:DSEG, CS:CSEG

MULD
PROC
FAR

MOV

AX, DW1L

MUL

DW2L

MOV

REZ+6, AX

MOV

REZ+4, DX

MOV

AX, DW1L

MUL

DW2H

ADD

REZ+4, AX

ADC

REZ+2, DX

ADC

REZ+0, 0H

MOV

AX, DW1H

MUL

DW2L

ADD

REZ+4, AX

ADC

REZ+2, DX

ADC

REZ, 0H

MOV

AX, DW1H

MUL

DW2H

ADD

REZ+2, AX

ADC

REZ, DX

RET
MULD
ENDP

CSEG
ENDS

END
14. Пример 6. Умножение двойных слов со знаком.

Простой переход к операциям IMUL невозможен, поскольку знак числа двойного слова зафиксирован только в старшем разряде двух двойных слов DW1H, DW2H.
Можно предложить следующий вариант решения:

1. Переводим отрицательные множители в положительные.

2. Параллельно определяем знак результата.

3. Выполняем умножение по готовой процедуре.

4. Корректируем результат с учетом знака.

Условимся, что исходные сомножители расположены в регистрах:

DX, CX – Mн.1

BX, AX – Mн.2

Выделим для анализа знака переменную ZNAK:

ZNAK = 0 ( REZ ( 0

ZNAK ( 0 ( REZ ( 0

Для перехода от отрицательного к положительному значению будем использовать вариант инвертирования и прибавления единицы.

Программная реализация

...................................
MOV

ZNAK, 0
; анализ знака Мн.1
CMP

DX, 0
JGE

MET1

NOT

DX

; Мн.1 = – Мн.1
NOT

CX

ADD

CX, 1
ADC

DX, 0
NOT

ZNAK

MET1:
CMP

BX, 0

; анализ знака Мн.2
JGE

MET2

NOT

BX

; Мн.2 = – Мн.2
NOT

AX

ADD

AX, 1
ADC

BX, 0
NOT

ZNAK

MET2:
MOV

DW1H, DX
; передача данных
MOV

DW1L, CX

MOV

DW2H, BX

MOV

DW2L, AX

CALL
MULD
MOV

DX, REZ
; возврат результата
MOV

CX, REZ+2

MOV

BX, REZ+4

MOV

AX, REZ+6

CMP

ZNAK, 0
; анализ знака результата
JNZ

MET3

NOT

DX

; REZ = – REZ
NOT

CX

NOT

BX

NOT

AX

ADD

AX, 1
ADC

BX, 0
ADC

CX, 0
ADC

DX, 0
MET3:
RET

15. Варианты адресации

Способы адресации можно условно разделить на три группы: регистровая, непосредственная и адресация к памяти. Третья группа включает целый ряд способов адресации: прямая, базовая, индексная, базово-индексная, а также перечисленные варианты со смещением.
1. Регистровая адресация – операнд (байт или слово) находится в регистре, например:

ADD AX


MUL
BX


MOV
CX, DX

2. Непосредственная адресация – операнд (байт или слово) является константой, например, число, адрес, символ ASCII:

MOV
AX, 0F7h

; пересылка числовой константы
MOV
BX, offset MAS
; адрес (смещение) переменной MAS
MOV
DL, 'A'

; в DL записывается код символа 'A'

3. Адресация к памяти, как правило, использует адресацию к сегменту данных или сегменту стека. Запись абсолютного адреса памяти содержит регистр сегмента, разделитель ":" и смещение, записанное как имя переменной или число в квадратных скобках, например:
DSEG

SEGMENT


P1
DB
1, 
2, 
3, 
4
; 4 байта




+0
+1
+2
+3
; смещение по байтам


P2
DW
5, 
6, 
7, 
8
; 4 слова




+4
+6
+8
+10
; смещение по словам
DSEG

ENDS
...............................
ASSUME
DS:DSEG, ...

...............................

MOV
AL, DS:[0] или MOV AL, DS:P1, AL = 1
MOV
AL, DS:[1] или MOV AL, DS:P1+1 или MOV AL, DS:P1[1], AL = 2

MOV
AX, DS:[4] или MOV AX, DS:P2, AX = 5
MOV
AX, DS:[6] или MOV AX, DS:P2+2 или MOV AX, DS:P2[2], AX = 6

Смещение всегда указывается в байтах, начиная с 0. Отсчет смещения начинается с начала сегмента или с переменной этого сегмента. При смещении по массиву байт прибавляется 1 (1 байт), по массиву слов – 2 (2 байта), по массиву двойных слов – 4 (4 байта). Размер выбираемых данных из памяти определяется размером регистра, например:


MOV
AL, DS:[0]
; из памяти выбирается 1 байт (по размеру AL)

MOV
AX, DS:[4]
; из памяти выбирается 2 байта (слово)

запись MOV AX, DS:[0] выберет из массива P1 2 байта, AX=21h (AH=2, AL=1), т.е. значение старшего адреса запишется в старший байт регистра AX.

Аналогично записывается адресация к сегменту стека: MOV AX, SS:[2].
Адресация к памяти без указания сегмента называется косвенной или относительной. Ниже приведены варианты косвенной адресации:
1. Прямая адресация – в качестве операнда выступает наименование переменной, объявленной в сегменте данных. При этом по умолчанию используется регистр DS, который должен быть настроен на сегмент данных с помощью оператора ASSUME, например:
DSEG

SEGMENT


P1
DB
1, 2, 3, 4


P2
DW
5, 6, 7, 8
DSEG

ENDS
...
ASSUME DS:DSEG

...
MOV
AL, P1
; AL = 1
ADD
DX, P2
; DX = 5
Прямая адресация со смещением:

MOV
AL, P1+2
; AL = 3

ADD
DX, P2+4
; DX = 7

или
MOV
AL, P1[2]
; AL = 3

ADD
DX, P2[4]
; DX = 7


2. Базовая и индексная адресация – для задания смещения в квадратных скобках кроме числовых констант могут использоваться регистры BX, BP, SI, DI (только эти регистры). Использование регистров BX, BP называется базовой адресацией, а регистров SI, DI – индексной адресацией. При абсолютной адресации эти регистры можно использовать в паре с сегментными регистрами DS, SS, ES, например, DS:[BP], SS:[DI], ES:[SI] и т.д. Если регистр сегмента явно не указан, то предполагается, что регистры BX, SI, DI используются с регистром DS (сегмент данных), а BP – с регистром SS (сегмент стека), например:

MOV
BX, offset P2
; BX = 4

MOV AX, [BX]

; AX = 5
при абсолютной адресации последняя запись выглядит так:


MOV AX, DS:[BX]
для стека:

MOV BP, SP
; пересылка указателя стека в регистр BP

MOV DX, [BP]
; эта запись аналогична MOV DX, SS:[BP]


Базовая и индексная адресация со смещением:
MOV AX, 3[BX]
MOV AX, [BX+3]
MOV AX, [BX]+3

эти выражения идентичны, при BX = 2 суммарное смещение равно 5.


3. Базово-индексная адресация. Относительный адрес определяется суммой содержимого базового и индексного регистров. Возможны следующие пары:

[BP][SI], 

[BP][DI], 

[BX][SI],

[BX][DI],

если в качестве базового регистра используется BP, то по умолчанию используется сегментный регистр SS (стек), если BX, то DS (данные). Порядок записи может быть обратный, т.е. сначала индексный регистр, а затем базовый. Например, для сегмента данных, приведенного выше, запишем выборку значений массива P2:

MOV CX, 4


; счетчик = 4

MOV BX, offset P2
; выборка смещения на P2


MOV DI, 0


; начальное смещение по P2 = 0
MET:
MOV AX, [BX][DI]
; AX = 5, 6, 7, 8

ADD DI,2


LOOP MET

Базово-индексная адресация со смещением:

MOV AX, 3[BX][DI]


MOV AX, [BX+3][DI]


MOV AX, [BX][DI+3]


MOV AX, [BX][DI]+3


MOV AX, [DI+BX+3]

эти выражения идентичны, при BX = 2, DI = 4, суммарное смещение равно 9.
16. Варианты обмена информации между процедурами

1. Использование аппарата EXTRN, PUBLIC. Здесь должны быть согласованы имена переменных.

PROC1




PROC2

EXTRN
B



EXTRN
A

PUBLIC
A



PUBLIC
B

Увязка имен реализуется на этапе компиляции программы.

2. Передача информации через регистры. Должны быть согласованы регистры.
PROC1




PROC2

MOV
AX, ...



ADD
AX, ...

CALL PROC2



---------------
MOV
BX, ...
MOV ..., BX



RET

Здесь обмен выполнен через регистры АХ, ВХ.

3. Передача информации через стек. Должна быть согласована структура стека.

PROC1


      
PROC2
PUSH Oп2

      

MOV
BP, SP

PUSH Oп1

      

MOV
AX, [BP+4]; Oп1
CALL PROC2
      

MOV
AX, [BP+6]; Oп2

	0
	2
	4
	6
	8
	10
	

	IP
	CS
	Оп1
	Оп2
	
	
	


4. Передача адреса информации через регистры. Передача адреса всегда эффективна, если передается большой объем информации в виде массивов или информационных структур.

PROC1



PROC 2

--------



---------

MOV   AX, seg M


MOV   ES, AX

MOV   BX, offset M

MOV   AL, ES:[BX+1]; элемент М[1]
CALL  PROC2


MOV   AH, ES:[BX+7]; элемент M[7]

	0
	1
	2
	3
	4
	5
	6
	7
	

	
	
	
	
	
	
	
	
	


Операция seg загружается значение регистра сегмента данных. Так как для абсолютного адреса здесь нельзя портить регистры сегментов CS, SS, DS, задействован дополнительный регистр сегмента ES.
5. Передача адреса информации через стек. Этот прием задействован в языках высокого уровня для определения входных и выходных параметров процедур.

PROC1



PROC2
FAR

--------



------------------

MOV   AX, seg M


MOV   BP, SP

PUSH  AX



MOV   AX, [BР+6] ;выборка адреса сегмента
MOV   AX, offset M

MOV   ES, AX

PUSH  AX



MOV   DI, [BP+4]   ;выборка смещения
CALL  PROC 2


MOV   AL, ES:[DI] ;выборка данных 

	0
	2
	4
	6
	8
	10
	

	IP
	CS
	off
	seg
	
	
	



Для сокращения записи текстов операндов можно использовать объявление EQU, например:
ADAN1
EQU
[BP+6]

ADAN2
EQU
[BP+4]

DAN

EQU
ЕS:[DI]
Здесь оператором EQU ставится соответствие сокращенного наименования длинного текста. Транслятор при просмотре программы автоматически заменяет сокращенные наименования на требуемые объявления. Используя указанные объявления текст PROC2 в последнем пункте имеет вид:

MOV

BР, SP


MOV

AX, ADAN1
MOV

ES, AX
MOV

DI, ADAN2
MOV

AL, DAN

17. Многофункциональная команда прерывания INT

INT тип прерывания


вектор прерываний

Программы обработки прерывания:

Команда INT реализует передачу управления к требуемой программе обработки прерывания. Адрес этой программы записан в векторе прерываний. Механизм передачи управления следующий:

1) регистр флагов RF, текущий адрес программы CS:IP (сегмент команд CS, счетчик команд IP) записываются в стек (запоминание состояния рабочей программы);

2) адрес из вектора прерываний записывается в CS:IP (передача управления программе обработки прерывания);

3) возврат из программы обработки прерывания выполняется командой IRET.
Выделяют следующие группы прерываний: внутренние, аппаратные, прерывания BIOS, прерывания DOS и др., например:

00h – внутреннее прерывание деления на ноль;

01h – внутренне прерывание пошагового режима;

05h – аппаратное прерывание клавиши PrnScr
08h – аппаратное прерывание системного таймера;

09h – аппаратное прерывание от клавиатуры;

0Bh – аппаратное прерывание от последовательного порта COM2;

10h – прерывание BIOS обслуживания видеосистемы;

13h – прерывание BIOS обслуживания дисков;

21h – диспетчер прерываний DOS и др.

Программы обработки прерываний могут иметь определенные соглашения по передаче информации через регистры. Например, для реализации ввода-вывода символов на экран требуется вызов прерываний DOS с номером 21h. В регистре АН нужно указать номер функции. Регистры DL и AL используются для размещения входного или выходного параметра. В DL находится символ, который должен быть выведен на экран, после вывода символа маркер сдвигается в следующую позицию автоматически. В АL размещается символ, который был введен с клавиатуры. Например:
АН = 7 – номер функции ввода символа с ожиданием
INT 21h
AL = ... – введенный символ

АН = 2 – номер функции вывода символа

DL = 20h – вывод пробела
INT 21h

АН = 2
DL = 8h – сдвиг маркера назад. 
INT 21h
18. Пример 7. Программы ввода-вывода символов на экран
Необходимо разработать две стандартные процедуры вывода на экран Wwrite, и ввода с клавиатуры Rread.

Любой символ в кодах ASCII представлен 1 байтом в диапазоне от 0...255 или 00…FF, например:
	 
	 0
	1
	2
	3
	4
	5
	6
	7
	8
	9
	A
	B
	C
	D
	E
	F

	0
	NUL
	SOH
	STX
	ETX
	EOT
	ENQ
	ACK
	BEL
	BS
	TAB
	LF
	VT
	FF
	CR
	SO
	SI

	1
	DLE
	DC1
	DC2
	DC3
	DC4
	NAK
	SYN
	ETB
	CAN
	EM
	SUB
	ESC
	FS
	GS
	RS
	US

	2
	 
	 !
	"
	#
	$
	 %
	&
	'
	(
	)
	*
	+
	,
	—
	.
	/

	3
	0
	1
	2
	3
	4
	5
	6
	7
	8
	9
	 :
	 ;
	<
	=
	>
	 ?

	4
	@
	A
	B
	C
	D
	E
	F
	G
	H
	I
	J
	K
	L
	M
	N
	O

	5
	P
	Q
	R
	S
	T
	U
	V
	W
	X
	Y
	Z
	[
	\
	]
	^
	_

	6
	`
	a
	b
	c
	d
	e
	f
	g
	h
	i
	j
	k
	l
	m
	n
	o

	7
	p
	q
	r
	s
	t
	u
	v
	w
	x
	y
	z
	{
	|
	}
	~
	DEL


00 - 1F – некоторые управляющие коды, среди которых можно выделить:

07h - звонок, 08h - маркер влево на 1 символ (клавиша BackSpace), 0Ah - маркер в начало строки, 0Dh - переход на следующую строку (клавиша Enter), 1Bh - клавиша Esc
20h - ' ',
21h - !,
22h - ",
23h - #, …
30h - 0,
31h - 1,
32h - 2,
33h - 3, …

41h - a,

42h - b,
43h - c,
44h - d, ...

80h - а,

81h - б,
82h - в,
83h - г, ...

Для расширения работы некоторые клавиши, связанные с определенным функциональным назначением, имеют расширенный код, который состоит из двух байт, причем младший всегда равен 0.

F1, F2 ... F12 - 3B00h, 3C00h, ... 4400h,

Home - 4700h,
( - 4800h,
Page Up - 4900h.

сочетания с функциональными клавишами также дает расширенный код:

F1 - 3B00h, Ctrl+F1 - 5E00h, Alt+F1 - 6800h, Shift+F1 - 5400h.
Реализация процедуры Wwrite

Пусть в управляющей процедуре, которой необходим вывод текстовой информации, есть строка: PR  DB
17,‘Введите строку : ’
; 17 – длина строки
Для реализации вывода этой строки будем использовать метод передачи адреса через стек, т.е. в управляющей процедуре имеем обращение:

MOV

AX, seg PR

PUSH
AX

MOV

AX, offset PR

PUSH
AX

CALL
Wwrite

Алгоритм
Установка адресных регистров
индекс = 1

<конец строки>
  Выход
выборка символа

вывод символа ( INT )

индекс + 1
Программная реализация

TITLE
PRIM 7

PUBLIC
Wwrite

A1
EQU
[BP+4]
; выбор смещения
A2
EQU
[BP+6]
; выбор сегмента
Size
EQU
ES:[DI]
; количество символов в строке 
Char
EQU
ES:[DI+BX]; выбор текущего символа
CSEG
SEGMENT

ASSUME
CS:CSEG
Wwrite
PROC
FAR

MOV

BP, SP

MOV

AX, A2

MOV

ES, AX

MOV

DI, A1

MOV

BX, 1
MOV

AH, 2

MET2:
CMP

BL, Size
JG

MET1

MOV

DL, Char

INT

21h
INC

BL

JMP

MET2

MET1:
RET
Wwrite
ENDP
CSEG
ENDS
END

Алгоритм процедуры Rread
Процедура должна обеспечивать ввод строки, например, с максимальным объемом 100 байт. При этом необходимо отфильтровать расширенные коды со звуковой сигнализацией, обеспечить редакцию строки по клавише BаckSpace и фиксировать конец ввода клавишей Enter. Передача информации о строке, куда должен производится ввод, реализуется вариантом записи адреса в стек (см. процедуру Wwrite). Алгоритм имеет вид:

установка адресных регистров, счетчик = 0





|
Next:


ввод символа




|



расширенный код ? (нет)




| (да)



звонок





|



сброс второго байта

Enter:


клавиша Enter ? (да)





| (нет)



клавиша BackSpace ? (нет)




| (да)


(да)
счетчик = 0 ?





| (нет)



строка[счетчик] = 0, счетчик – 1,




маркер (, вывод '_', маркер (
Lim:

(да)
предел длины строки ?




| (нет)




счетчик +1,




строка[счетчик] = символ,




эхо-вывод

Fiks:


фиксация размера строки
В управляющей процедуре должна быть рабочая строка. Например, string DB 101 DUP(0). При взаимодействии предусматривается выход по клавише Enter и редакция по клавише BackSpace. Рассмотрим ряд фрагментов реализации:

MOV
AH,7
;ввод символа с ожиданием

INT
21h


...
AL
;введенный символ в AL


MOV
AH,2
;вывод символа (звонок)

MOV
DL,7


INT
21h

MOV
AH,2
;сдвиг маркера влево

MOV
DL,8

INT
21h
Программная реализация
TITLE
PRIM7
PUBLIC
Rread
DSEG
SEGMENT

Limit
db 100


;предел длины строки = 100
DSEG
ENDS

CSEG
SEGMENT

ASSUME
CS:CSEG, DS:DSEG
A1
EQU [BP+4]

A2
EQU [BP+6]

Size
EQU ES:[BX]

Char
EQU ES:[BX+DI]

Rread
PROC
FAR
MOV

BP, SP
;установка адресных регистров
MOV

AX, A2

MOV

ES, AX

MOV

BX, A1

MOV

DI, 0

;счетчик = 0
Next:

MOV

Ah, 7

;ввод текущего символа
INT

21h
CMP

AL, 0

;проверка расширенного кода
JNE

Enter
MOV

Ah, 2

;звонок
MOV

DL, 7

INT

21h
MOV

Ah, 7

;сброс второго байта (повтор ввода)
INT

21h

JMP

Next
Enter:

CMP

AL, 13
;клавиша Enter ?
JE

Fiks
CMP

AL, 8

;клавиша BackSpace
JNE

Lim

CMP

DI, 0

;счетчик = 0 ?

JE

Next

MOV

Char, 0
DEC

DI

MOV

Ah, 2
MOV

DL, 8

;сдвиг маркера влево
INT

21h
MOV

DL, 20
;вывод пробела
INT

21h
MOV

DL, 8

;сдвиг маркера влево
INT

21h

JMP

next

Lim:

CMP

DI, Limit
JE

Next

INC

DI

;счетчик+1
MOV

Char, AL

MOV

AH, 2

MOV

DL, AL
;эхо-вывод
INT

21h

JMP

Next

Fiks:

MOV

AX, DI

MOV

Size, AL

RET

Rread
ENDP
CSEG
ENDS
END
Программы Rread, Wwrite являются стандартными и могут быть использованы в разных функциональных задачах.

Управляющая процедура ввода-вывода может быть выполнена в последовательности: вывод подсказки – ввод строки – вывод результата набора.

TITLE
MAIN7

EXTRN
Rread:  FAR, Wwrite: FAR;

stEk

Segment
STACK
db
64 dup (?)

stEk

ends

DSEG
segment

String
db
101 dup (0)

PR
db
17,'введите строку : '     

RS
db
19,0Dh,0Ah,'результат ввода : '
;0Dh – перевод строки, 0Аh – возврат каретки

dSEG
ends

cSEG
segment

Assume
SS:stEk, ds:dSEG, cs:cSEG

Start:
MOV

AX, DSEG

;загрузка сегмента данных
MOV

DS, AX
MOV

AX, seg PR

;вывод приглашения
PUSH
AX
MOV

AX, offset PR

PUSH
AX
CALL
Wwrite

POP

AX

POP

AX

MOV

AX, seg String
;ввод строки
PUSH
AX
MOV

AX, offset String
PUSH
AX
CALL
Rread
POP

AX

POP

AX

MOV

AX, seg RS

;вывод приглашения для результата
PUSH
AX
MOV

AX, offset RS
PUSH
AX
CALL
Wwrite

POP

AX

POP

AX

MOV

AX, seg String
;вывод результата набора
PUSH
AX
MOV

AX, offset String
PUSH
AX
CALL
Wwrite
POP

AX

POP

AX


MOV
AX, 4C00h


;завершение программы

INT
21h
CSEG
ENDS
END
Start
В результате работы программы, например, при вводе строки 'PRIM' на экране появится информация:

Введите строку : PRIM
Результат ввода : PRIM
19. Реализация вызова процедур Assembler на языке Pascal

Допустим, в одном проекте часть программы написана на языке низкого уровня (Assembler), например, чтение информации с портов, взаимодействие с контроллерами, процедуры доступа к данным и др., а интерфейс программы реализован на языке высокого уровня (например, Pascal). Необходимо уметь программировать обращения к процедурам Assembler из процедур Pascal. Реализуем эту задачу с процедурами Rread и Wwrite (реализация управляющей процедуры на языке Pascal). Будем использовать переменную типа pointer – указатель (физический адрес переменной), а также загрузку этих типов переменных операций.

Программная реализация

program Main7;
const

PR : string[17] = 'Bведите строку : ';

RS : string[20] = #13, #10,'Pезультат ввода : ';

var

st : string[101];

p : pointer;

procedure Rread (p : pointer); external;

procedure Wwrite (p : pointer); external;
{$L Rread.obj}

{$L Wwrite.obj}

begin

st := '';
{Очистка строки результата}

p := @PR;
Wwrite(p);

p := @st;
Rread(p);

p := @RS;
Wwrite(p);

p := @st;
Wwrite(p);

end.

Примечание:
1. Данная управляющая программа полностью выполняет функции ранее рассмотренной управляющей процедуры на языке Assembler.

2. Процедуры Rread и Wwrite включаются в состав программы Main7 из объектных файлов (*.obj), т.е. уже после их трансляции. После объявления процедур Rread и Wwrite ставится служебное слово external (внешний).
3. {$L Rread.obj} – директива включения объектного файла в программу.
4. Выражение p := @ST присваивает адрес переменной ST (аналог seg, offset в Assembler).

20. Расчет установившегося режима локальной разомкнутой электрической сети 10 кВ
Имеется локальная разомкнутая электрическая схема, по которой поступают телеизмерения U, I, P. Ставится задача расчета установившегося режима этой схемы с использованием контроллера МП.

Расчетная модель электрической сети представляется в виде упорядоченного набора его элементов (узел + питающая ветвь). Элементы упорядочены в последовательности расчета токораспределения, соответствующего порядку свертки элементов при расчете токораспределения. Прямым циклом по элементам реализуется токораспределение, обратным циклом – режим напряжений. Будем использовать упрощенную модель электрической сети:

где
     N
– номер узла;

       P, U, J
– соответственно заданная мощность и рассчитанные напряжение и ток узла;

            R, I
– заданное сопротивление и рассчитанный ток питающей ветви;

                S 
– ссылки на питающие элементы.
По сравнению с известными моделями разомкнутых сетей для упрощения исключили вектор упорядочения, т.е. участки будем записывать в порядке нумерации. Все параметры и режимные характеристики имеют размерность слова и записаны в массиве SHEMA.

SHEMA
DW
P1, R1, U1, J1, I1,
P2, R2, U2, J2, I2
и т.д.





1


2
Пример:
Пусть исходные данные для сети имеют вид:

SHEMA
DW
50,3,0,0,0,
30,4,0,0,0,
25,2,0,0,0,
0,0,110,0,0



1

2

3

4

Этим данным соответствует схема:

Начальные адреса каждого элемента в массиве SHEMA связаны с номером узла формулой: Адрес = (N–1)*10. Адреса элементов идут в последовательности 0, 10, 20, 30, …
Ссылки на питающие элементы будем хранить в отдельном массиве S. Для приведенной 4-узловой схемы массив S выглядит так:
S
DW
20, 20, 30

Для схемы из девяти узлов


S
DW 60, 60, 50, 50, 50, 70, 70, 80

Для обращения к элементам рассмотренной модели будем пользоваться базово-индексной адресацией: Δ[BX][SI],

где
Bx – базовый  регистр,  содержащий  начальный  адрес  массива SHEMA;

SI  – индекс    смещения    к    необходимому    элементу массива SHEMA;
Δ   – смещение по записи.
	
	P
	R
	U
	J
	I

	∆
	0
	2
	4
	6
	8


например:
4[BX][SI] → адресация к напряжению;

8[BX][SI] → адресация к току участка;

0[BX][SI] → адресация к мощности участка, и т.д.;
Переход к следующему элементу: SI + DELTA, где DELTA = 10.

Алгоритм расчета режима


В алгоритме выделим четыре последовательных шага:

1. Загрузка начальных напряжений;
2. Расчет токов узлов и начальная загрузка токов линий;

3. Расчет токораспределения;

4. Расчет режима напряжения;

5. Контроль сходимости итерационного процесса.

Для упрощения итерационный процесс будем завершать после выполнения заданного количества итераций.

Например, 1-я итерация для 4-узловой схемы выглядит так:



1) Загрузка U

2)  J = P / U
    I = J

3) IП = IП + I

4) U = UП – R*I

Рассмотрим алгоритмы реализации этих шагов:

1. Загрузка начальных напряжений. Организуем цикл по элементам k раз (k – количество элементов) и везде запишем напряжение балансирующего узла (это можно сделать для схем одного уровня напряжений):
SI = 0, CX = k
4[BX][SI] = UB

SI + DELTA
; DELTA = 10
LOOP

2. Расчет токов узлов (I=P/U) и начальная загрузка токов линий. Имеем аналогичный предыдущему цикл:
SI = 0, CX = k
6[BX][SI] = 0[BX][SI] / 4[BX][SI]

8[BX][SI] = 6[BX][SI]

SI + DELTA

LOOP

3. Расчет токораспределения. Здесь надо использовать два индексных регистра:

SI – текущий элемент;

DI – питающий элемент.
Цикл должен выполняться длиной k–1. Значение DI рассчитывается через массив ссылок S: DI = S[2*SI / DELTA], SI = 0, 10, 20, ..., DI = S[0], S[2], S[4], ... 

CX = k–1, SI = 0

DI = S[2*SI/DELTA]

8[BX][DI] = 8[BX][DI] + 8[BX][SI]

SI + DELTA
LOOP

4. Расчет режима напряжений. Здесь цикл должен быть организован в обратном направлении. Как и в п.3 SI – текущий элемент, DI – питающий элемент. Цикл k–1 раз, поскольку напряжение балансирующего узла не меняется. Реализуется выражение U = UП – R*I, где UП – напряжение питающего элемента.
CX = k–1


;CX 
= ...
5,
4,
3,
2,
1

SI = (CX–1)*DELTA
;SI 
= ...
40,
30,
20,
10,
0

DI = S[(CX–1)*2]

;DI
= ...
S[8],
S[6],
S[4],
S[2],
S[0]
4[BX][SI] = 4[BX][DI] – 2[BX][SI] * 8[BX][SI]
      U

      UП

R

I
LOOP

Во всех рассмотренных алгоритмах необходимо решать вопрос о масштабировании физических параметров и расчетных характеристиках.

21. Целочисленное программирование

В системе команд МП отсутствуют команды с действительными числами, которыми являются P, R, U, J, I. Будем использовать аппарат целочисленной арифметики, называемый фиксированной точкой. Аппарат предполагает следующее:

1) Выбор для каждой физической величины своего фиксированного масштаба, определяемого инженерной точностью, например:

P
→ точность

0,1 МВт
→ 
М = 101

R
→ точность

0,01 Ом
→ 
М = 102

U
→ точность

0,01 кВ
→ 
М = 102

J, I
→ точность

0,001 кА
→ 
М = 103

Выбранные масштабы не обязательно должны быть кратными 10. 
2) Выбор размера ячейки памяти по максимально допустимому значению каждой физической величины.

P
→ максимум
99,9 МВт → 
999 ф.т.
→ 
слово

R
→ максимум
99,99 Ом → 
9999 ф.т.
→ 
слово

U
→ максимум
99,99 кВ → 
9999 ф.т.
→ 
слово

J, I
→ максимум
9,999 кА → 
9999 ф.т.
→ 
слово

3) Выбор уравнительных масштабов для всех формул, реализующих действие над физическими величинами.

Формула:
I = P/U,

MI = MУ*MP/MU,

MУ = (MI*MU)/ MP = (103*102)/101 = 104.

Формула:
U = U – I*R,

MU= MU – MУ*MI*MR,

MУ = MU /(MR*MI) = 102 /(102*103) = 10–3.

При получении масштаба меньше 1 в реализации следует принимать масштаб с положительной степенью, но в реализациях делить на этом масштаб.

4) Необходимо определить последовательность операций, в каждой конкретной формуле, чтобы обеспечить максимальную точность результата и минимум затрат программной реализации.

Возможно ступенчатое масштабирование, например:

X = A·B·C·D·E·F·MУ



в этом случае MУ = MУ(1) ·MУ(2) ·MУ(3)…

В нашем конкретном примере будем выполнять операции в следующей последовательности:
I = P·MУ / U 

U = R*I / MУ*, MУ* = 103.
Для упрощения в реализации не учитываем округления последнего знака, который выходит за разрядную сетку.

22. Пример 8. Реализация расчета установившегося режима локальной разомкнутой электрической сети 10 кВ
Данные массива SHEMA установлены для следующего примера схемы:


TITLE
PRIM8

PUBLIC
REGIM

DSEG
SEGMENT

k

DW 4



;количество узлов
ki

DW 10


;количество итераций
DELTA
DW 10


;шаг по массиву SHEMA
SHEMA
DW 
5, 150, 0, 0, 0,
;параметры схемы
DW
31, 321, 0, 0, 0,
;с учетом исходных
DW
42, 85, 0, 0, 0,
;масштабов
DW
0, 0, 0, 0, 0

S

DW
20, 20, 30
;массив ссылок на питающие элементы
Ub

DW
11000

;напряжение балансирующего узла
MY1

DW
10000

;уравнительный масштаб 1
MY2

DW
1000

;уравнительный масштаб 2
DSEG

ENDS

CSEG

SEGMENT

ASSUME
CS:CSEG, DS:DSEG

REGIM
PROC
FAR

MOV

AX, DSEG

;загрузка DS
MOV

DS, AX

MOV

BX, offset SHEMA

XOR

SI, SI

MOV

CX, k


;загрузка начальных U



MOV

AX, Ub

MET1:
MOV

4[BX][SI], AX
ADD

SI, DELTA

LOOP
MET1

MOV

CX, ki

;организация итерационного цикла
MET2:
PUSH
CX
XOR

SI, SI


;расчет токов узлов
MOV

CX, k
MET3:
MOV

AX, 0[BX][SI]

MUL

MY1

DIV

WORD PTR 4[BX][SI]

MOV

6[BX][SI], AX

MOV

8[BX][SI], AX

ADD

SI, DELTA

LOOP
MET3

XOR

SI, SI


;токораспределение
MOV

CX, k
DEC

CX

MET4:
MOV

AX, SI

SHL

AX, 1
DIV

DELTA

MOV

DI, AX

MOV

DI, S[DI]

MOV

AX, 8[BX][DI]

ADD

AX, 8[BX][SI]

MOV

8[BX][DI], AX

ADD

SI, DELTA

LOOP
MET4

MOV

CX, k


;режим напряжений
DEC

CX

MET5:
MOV

SI, CX

DEC

SI

MOV

DI, SI

MOV

AX, SI

MUL

DELTA

MOV

SI, AX

SHL

DI, 1

MOV

DI, S[DI]

MOV

AX, 2[BX][SI]

MUL

WORD PTR 8[BX][SI]

DIV

MY2

MOV

DX, 4[BX][DI]

SUB

DX, AX

MOV

4[BX][SI], DX

LOOP
MET5

POP

CX


;переход к следующей итерации
LOOP
MET2
RET
REGIM
ENDP

CSEG
ENDS

END

REGIM
23. Пример 9. Программа умножения матриц
1. Стандартное алгоритмическое решение

организация циклов:

i=1÷n

l=1÷k

j=1÷m

C[i, l] = C[i, l] + A[i, j]*B[j, l]

2. Программное решение:

· числа принимаем размером в слово;

· матрицы будем размещать по одному принципу (по строкам);

· арифметику выполняем в режиме фиксированной точки, где MA, MB, MC – соответственно масштабы матриц А, В, С;
· MY – уравнительный масштаб: MA*MB*MY = MC ( MY=MC/(MA*MB).

Пример:

Пусть МА=101, МВ=103, МС=102, тогда МY=10-2,

	
	
	
	
	
	
	
	
	
	
	+0
	+2
	+4
	+6

	
	
	2.1
	4.7
	6.5
	7.3
	
	MATA
	DW
	+0
	21,
	47,
	65,
	73,

	A
	=
	8.2
	1.3
	7.4
	2.2
	
	
	
	+8
	82,
	13,
	74,
	22,

	0.111
	0.125
	
	
	MATB
	DW
	+0
	2,
	111,
	125,
	

	B
	=
	0.346

	0.005

	346,
	42,
	5,

	DW

	=

C11=21*2 + 47*346 + 65*0 + 73*1245 = 107189/102 = 1072 = 10,72
3. Реализация вложенных циклов

CX=N

M3:

PUSH CX

CX=K

M2:

PUSH CX



CX=M
M1:

...

LOOP M1

POP CX
LOOP M2

POP CX
LOOP M3


Вложенные циклы реализуются за счет сохранения текущего индекса каждого цикла (регистр СХ) в стеке. Перед оператором LOOP текущий индекс выгружается из стека, а при переходе по метке цикла снова загружается в стек.
4. Реализация адресации
	MATA[BX, DI]
	
	DI

	J
	(
	
	
	

	
	
	
	+0
	+2
	+4
	

	J = 0, 2, 4, 6, ... 16
	
	
	+6
	+8
	+10
	

	
	
	
	+12

В матрицах А, В используется базово-индексная адресация. Базовые регистры фиксируют текущую строку и столбец в матрицах А и В. Индексные регистры реализуют движение по текущей строке и столбце матриц А, В. 
В матрице С адреcация реализована по одному индексу, поскольку последовательность размещения элементов массива в памяти не требует переходов по столбцам и строкам.
5. Общий алгоритм

Размерность матриц А, В (N x M и M x K) будем хранить в переменных N, M, K. Если элементы матриц имеют размерность слово (2 байта), то переходы между строками матриц А и В имеют вид:


BX = BX + 2*M (матрица А);

SI = SI + 2*K (матрица B).


Регистры DI, BP и переменная J изменяются с шагом 2.


J = 0, BX = 0, 

CX = N

M3:
PUSH CX


BP = 0, 

CX = K

 M2:
PUSH CX


SI = 0, DI = 0, Sum1 = 0, Sum2 = 0, 

CX = M
   M1:
AX, DX = MATA[BX,DI] * MATB[BP,SI]


Sum1, Sum2 = Sum1, Sum2 + AX, DX


DI+2, SI+2*K


LOOP M1


Sum1, Sum2 / MY ( AX ( MATC[J]


J+2, BP+2


LOOP M2


BX + 2*M


LOOP M3

6. Программная реализация
Наименования и размерность матриц являются внешними именами. Внутренние переменные: J, Sum1, Sum2. Умножение и деление выполняется с учетом знаков (команды IMUL, IDIV).
TITLE
PRIM10
EXTRN
MATA:WORD, MATB:WORD, MATC:WORD, 

N:WORD, M:WORD, K:WORD, MY:WORD,

PUBLIC
MULM
DSEG
SEGMENT

J
DW
Sum1
DW
Sum2
DW
DSEG
ENDS

CSEG
SEGMENT

ASSUME
CS:CSEG, DS:DSEG

MULM
PROC
FAR

XOR
AX, AX

MOV
J, AX

;J = 0 (10 тактов), MOV J,0 (10+EA (234) тактов)
XOR
BX, BX
;BX = 0
MOV
CX, N

M3:

PUSH CX
XOR
BP, BP
;BP = 0
MOV
CX, K

M2:

PUSH CX
XOR
SI, SI

;SI = 0
XOR
DI, DI
;DI = 0
XOR
AX, AX

MOV
Sum1, AX
;Sum1 = 0
MOV
Sum2, AX
;Sum2 = 0
MOV
CX, M

M1:

MOV
AX, MATA[BX][DI]
IMUL  MATB[BP][SI]
; MATA[BX][DI]*MATB[BP][SI] ( AX,DX
ADD
Sum1, AX

ADC
Sum2, DX

INC
DI
INC
DI

;DI+2
MOV
AX, K
SHL
AX, 1

;2*K
ADD
SI, AX
;SI + 2*K
LOOP M1

MOV
AX, Sum1

MOV
DX, Sum2

IDIV
MY

;Sum1, Sum2 / MY ( AX
MOV DI, J

MOV
MATC[DI], AX

INC
DI

INC
DI

;J+2
MOV  J, DI

INC
BP

INC
BP

;BP+2
POP
CX

LOOP  M2

MOV
AX, M
SHL
AX, 1

;2*M

ADD
BX, AX
;BX + 2*M
POP
CX

LOOP  M3

RET
MULM
ENDP
CSEG
ENDS

END
В определенных вариантах умножения матриц нельзя заранее определить уравнительный масштаб, а в общем случае и масштабы для конкретных матриц. В этих случаях необходим переход к числам с плавающей точкой.

24. Разработка формата чисел с плавающей точкой

Для реализации формата необходимо согласовать структуру составного информационного слова, в котором присутствуют следующие компоненты:

· Мантисса числа;

· Порядок числа;

· Знак порядка;

· Знак мантиссы.

Мантиссу будем представлять шестнадцатиричным нормализованным числом (после точки обязательно значащая цифра), например:

AB1.CDE37 ( 0.AB1CDE37*163

0.002FFE3 ( 0.2FFE3*16–2

Порядок также должен быть шестнадцатиричным числом. Для записи знаков выделим по одному разряду. Число с плавающей точкой будем хранить в двойном слове с следующей структурой:

	ZC
	ZP
	P1
	P2
	
	C1
	C2
	
	C3
	C4
	
	C5
	C6


ZC, ZP – биты знаков мантиссы и порядка (0 - "+", 1 - "–");
P1, P2 – шестнадцатеричные цифры порядка. Максимальный порядок равен 3F (в двоичном виде – 0011 1111);

C1 - C6 – шестнадцатиричные цифры мантиссы.


Для перевода десятичных чисел с плавающей точкой в шестнадцатиричные необходимо выполнить шаги:


1. Перевод целой части 10 ( 16.


2. Перевод дробной части 10 ( 16.


3. Нормализация шестнадцатиричного числа и расчет порядка.


4. Запись числа с плавающей точкой в виде шестнадцатиричной константы: _ _ _ _ _ _ _ _ h (8 цифр).


В общем случае шестнадцатиричное число в позиционной системе счисления представляется:


Ц2Ц1Ц0 . Д1Д2Д3 = Ц0*160 + Ц1*161 + Ц2*162 + Д1*16–1 + Д2*16–2 + Д3*16–3

целая
     дробная


часть      часть


Перевод дробной части реализуется методом последовательного умножения на 16 по следующей схеме:


Д0*16 ( Ц1 + Д1

Д1*16 ( Ц2 + Д2

Д2*16 ( Ц3 + Д3

…………………


результат перевода: 0.Ц1Ц2Ц3Ц4Ц5…
Примеры:

1. 998710 ( 270316 ( 0.2703*164 ( 04270300h

2. –998710 ( –0.2703*164 ( 

84270300h

3. 21.310 (?16
 
2110 ( 1516

перевод целой части


0.3*16 = 4.8

перевод дробной части


0.8*16 = 12.8



0.8*16 = 12.8



0.8*16 = 12.8



...



0.310 ( 0.4CCCCC...16


21.310 (15.4CCCCC...16 ( 0.154CCCCC...*162 ( 02154СССh

4. 0.00710 (?16


0.007*16 = 0.112



0.112*16 = 1.792



0.792*16 = 12.672



0.672*16 = 10.752



0.752*16 = 12.032



0.032*16 = 0.512



0.512*16 = 8.192



0.192*16 = 3.072



...



0.007 ( 0.01CAC083 ( 0.1CAC083*16–1 ( 411CAC08h
выполним контроль правильности перевода используя 
позиционную систему:



0.1CAC08316 ( ?10 ( 1/16 + 12/162 + 10/163 + 12/164 + 0/165 + 8/166 + 3/167 = 0.0625 + 0.046875 + 0.00244140625 + 0.00018316 + ... = 0.11181640625

0.11181640625*16–1 ( 0.007


5. –0.00710 (C11CAC08h
25. Пример 10. Умножение чисел с плавающей точкой

В общем случае умножение выполняется по выражению:

М1*16P1 * М2*16P2 = МR*16PR, где

МR = М1* М2

PR = P1 + P2 + PH

PH – порядок нормализации произведения мантисс МR, может принимать значения 0 или -1, например:


0.1A*162 * 0.2B*162 = 0.045E*164 = 0.45E*163 (PH = -1)

0.1A*162 * 0.E5*162 = 0.1742*164 (PH = 0)


Поскольку мантиссы двух множителей содержатся в двойных словах, будем использовать процедуру умножения двойных слов без знака MULD (пример 5). Напомним, что исходные множители для этой процедуры записываются в слова DW1L, DW1H, DW2L, DW2H, а результат в поле REZ длиной 4 слова.


Процедуру умножения чисел с плавающей точкой назовем MULF. Исходные множители и результат находятся переменных M1, M2, REZF размером двойное слово.


Отметим следующие особенности реализации:


1) Пересылка M1, M2 ( DW1L, DW1H, DW2L, DW2H

M1 (M2)




представление в памяти
	DD
	4
	3
	2
	1
	(
	1
	2
	3
	4



	DW
	0
	3
	
	2
	1
	(
	1
	2
	
	3
	0



     DWH

 DWL


  DWL

DWH
Байты пронумерованы по старшинству 1, 2, 3, 4. Значение мантиссы расположено в байтах 1 – 3, старший байт (4-й) пересылать не нужно, так как он содержит порядок числа и два знака (мантиссы и порядка). Поэтому старший байт DW1H после пересылки равен 0. В памяти компьютера слова и двойные слова сохраняются по принципу "младший байт ( младший адрес", например:

M1 = 02 15 4C CC ( в памяти CC 4C 15 02

после пересылки:

DW1L = 4C CC ( в памяти CC 4C
DW1H = 00 15 ( в памяти 15 00

пересылка множителей, например М1 ( DW1L, DW1H, реализуется командами:

MOV  BX, offset M1


MOV  AX, [BX]


MOV  DW1L, AX


MOV  AX, [BX+2]


AND  AX, 0FFh
;обнуление 4-го байта

MOV  DW1H, AX

2) Обратная пересылка REZ ( REZF
REZ
DW (результат умножения в массиве из 4 слов)

	+0
	7
	8
	+2
	5
	6
	+4
	3
	4
	+6
	1
	2



	REZF DD (дв. слово)
	1
	2
	3
	4


Результат умножения после процедуры MULD записывается в структуру REZ. Первое слово структуры REZ (байты 7,8) содержит нули, поскольку произведение 3 байта * 3 байта = максимум 6 байт, например, при перемножении максимальных чисел имеем результат:

00 FF FF FF * 00 FF FF FF = 00 00 FF FF FE 00 00 01  – два старший байта = 0

Байты 5,6 структуры REZ нужно переслать в байты 2,3 переменной REZF, а 4-й байт REZ в 1-й байт REZF. Пересылка реализуется командами:


MOV  BX, offset REZF

MOV  AX, REZ+4


MOV  [BX], AH
;заполнение 1 байта REZF

MOV  AX, REZ+2


MOV  [BX+1], AX
;заполнение 2,3 байтов REZF
Произведение минимальных чисел дает результат:

10 00 00 * 10 00 00 = 01 00 00 00 00 00  – старшая цифра мантиссы равна 0, поэтому результат необходимо нормализовать (сдвинуть влево на 1 цифру или 4 бита). Таким образом необходимо проверять на ноль старшую цифру 5-го байта структуры REZ, и если ноль, то выполнять сдвиг влево на 4 бита. 


3) Нормализация структуры REZ
REZ          +2                    +4                    +6

	
	
	
	
	
	
	
	
	
	
	
	
	
	
	
	
	
	
	


                      F  F  F  0         F  0  0  0   AND

пересылка тетрады
                                OR        *          *   SHR (12 бит)
между словами

Степень нормализации (Рн) будем хранить в регистре DL. Смещение по структуре REZ будем хранить в регистре SI. Начальное значение SI = 2. Операцию сдвига будем выполнять для слов REZ[SI]. Операция пересылки тетрады между словами выполняется командами:


MOV  AX, REZ[SI+2]
;выборка след. слова в AX

AND  AX, 0F000h

;выделение старшей тетрады

MOV  CL, 12


;сдвиг старшей тетрады

SHR  AX, CL


;в младшую

OR   REZ[SI], AX

;пересылка тетрады

Полная реализация сдвига структуры REZ показана ниже:
        XOR DL,DL   ;Pн = 0

        MOV SI,2    ;начало с второго слова REZ
        MOV AX,REZ[SI]

        AND AH,0F0h ;старшая цифра = 0 ?
        JNZ Round   ;если нет, то идем на округление
        DEC DL      ;Рн = -1

Norm:   MOV CL,4

        SHL REZ[SI],CL ;сдвиг влево очередного слова
        CMP SI,6    ;если SI = 6 (последнее слово)

        JE  Round   ;то идем на округление
        MOV AX,REZ[SI+2]   ;переносим тетраду
        AND AX,0F000h      ;между словами REZ
        MOV CL,12

        SHR AX,CL

        OR  REZ[SI],AX
        INC SI             ;переход к следующему слову
        INC SI

        JMP Norm

Round:  ...

4) Округление результата
REZ                        мантисса
	
	
	
	
	+2
	2
	1
	+4
	4
	3
	+6
	
	



                                            * * * * * * * *
                                                  прибавить к мантиссе 

Для округления результата нужно прибавить к мантиссе старший бит тетрады, следующей за мантиссой:

Round:  MOV AX,REZ[4]

        SHL AL,1

;заполнение CF старшим битом 

        JNC CalcP
;если CF = 0, то идем дальше
        ADC AH,0     ;прибавление CF к мл.байту мантиссы
        ADC REZ[2],0 ;прибавление CF к ст.слову мантиссы
        MOV REZ[4],AX;возврат мл.байта мантиссы в REZ
CalcP:  ...


5) Расчет знака и порядка результата: PR =  (    PМ1   (   PМ2  +  PH
                                                                             AL   BH  AH  BL  AL      DL
                  BX+3

M1                                           M2                                           REZF

	1
	2
	3
	4
	*
	1
	2
	3
	4
	=
	1
	2
	3
	4


                        3F – порядок (степень)        80 – знак числа
                        40 – знак порядка
CalcP:  MOV BX,offset M1

        MOV AH,[BX+3]   ;старший байт М1 в AH
        MOV BX,offset M2

        MOV AL,[BX+3]   ;старший байт М2 в AL
;расчет знака результата ( DH
        MOV DH,AH
        XOR DH,AL
        AND DH,80h      ;знак результата
        MOV BH,AH
        AND AH,3Fh      ;выборка порядка Pm1

        AND BH,40h      ;выборка знака Pm1

        JZ  NextM       ;если знак "+", то дальше
        NEG AH          ;если нет, то меняем знак Pm1
NextM:  MOV BL,AL

        AND AL,3Fh      ;выборка порядка Pm2

        AND BL,40h      ;выборка знака Pm2

        JZ  SumP        ;если знак "+", то дальше
        NEG AL          ;если нет, то меняем знак Pm2

SumP:   ADD AL,AH       ;Pr = Pm1 + Pm2 ( AL
        ADD AL,DL       ;Pr = Pr + Pн ( AL
        JNS WrREZF     ;если Pr не < 0, то запись REZF 

        NEG AL         ;иначе меняем знак Pr
        OR  AL,40h     ;и устанавливаем бит знака порядка
WrREZF: ...

Шаги программной реализации:


1. Пересылка множителей М1, М2 ( DW1L, DW1H, DW2L, DW2H
2. Вызов процедуры MULD
3. Нормализация результата (структура REZ)
4. Округление результата (структура REZ)
5. Расчет знака (DH) и порядка результата (AL)
6. Пересылка результата REZ ( REZF
TITLE   MULF

PUBLIC  M1,M2,REZF,MULF

EXTRN DW1L:WORD,DW1H:WORD,DW2L:WORD,DW2H:WORD,REZ:WORD,MULD:FAR

DSEG    SEGMENT PARA PUBLIC

    M1      DD

    M2      DD

    REZF    DD       

DSEG    ENDS         

CSEG    SEGMENT PARA PUBLIC

    ASSUME  CS:CSEG,DS:DSEG

    MULF    PROC    FAR

; --------------------------- обнуление REZF
        XOR AX,AX

        MOV BX, offset REZF

        MOV [BX],AX

        MOV [BX+2],AX

;--------------------------- M1,M2 -> DW1L,DW1H,DW2L,DW2H

        MOV BX,OFFSET M1

        MOV AX,[BX]

        MOV DW1L,AX

        MOV AX,[BX+2]

        AND AX,0FFh

        MOV DW1H,AX

        MOV BX,OFFSET M2

        MOV AX,[BX]

        MOV DW2L,AX

        MOV AX,[BX+2]

        AND AX,0FFh

        MOV DW2H,AX

        CALL MULD

;---------------------------Нормализация REZ
        XOR DL,DL 

        MOV SI,2

        MOV AX,REZ[SI]

        AND AH,0F0h

        JNZ Round
        DEC DL

Norm:   MOV CL,4

        SHL REZ[SI],CL

        CMP SI,6

        JE  Round

        MOV AX,REZ[SI+2]

        AND AX,0F000h

        MOV CL,12

        SHR AX,CL

        OR  REZ[SI],AX

        INC SI

        INC SI

        JMP Norm

;---------------------------Округление REZ

Round:  MOV AX,REZ[4]

        SHL AL,1

        JNC CalcP

        ADC AH,0

        ADC REZ[2],0

        MOV REZ[4],AX

;---------------------------Расчет Pr = Pm1 + Pm2 + Pn

CalcP:  MOV BX,offset M1

        MOV AH,[BX+3]   ;Pm1 -> AH

        MOV BX,offset M2

        MOV AL,[BX+3]   ;Pm2 -> AL

        ;-------------------знак результата -> DH
        MOV DH,AH
        XOR DH,AL

        AND DH,80h
        ;-------------------расчет порядка -> AL
        MOV BH,AH
        AND AH,3Fh     

        AND BH,40h     
        JZ  NextM      

        NEG AH

NextM:  MOV BL,AL

        AND AL,3Fh     

        AND BL,40h     

        JZ  SumP       

        NEG AL

SumP:   ADD AL,AH      

        ADD AL,DL      

        JNS WrREZF

        NEG AL

        OR  AL,40h     

;-------------------------- пересылка REZ -> REZF 

WrREZF: OR  AL,DH       ;добавление знака результата в AL
        MOV BX,offset REZF
        MOV [BX+3],AL   ;запись старшего байта AL -> REZF
        MOV AX,REZ[4]

        MOV [BX],AH

        MOV AX,REZ[2]

        MOV [BX+1],AX

        RET

    MULF    ENDP

CSEG    ENDS

END        

Пример:

M1 = C4 1A 56 08 (–1.569751475472*10–6),M2 = 03 10 FE A1 (271.91430664)

DW1H = 00 1A, DW1L = 56 08, 

DW2H = 00 10, DW2L = FE A1

REZ = 00 00 01 BF 92 6C 0B 08

Нормализация: 
REZ = 00 00 1B F9 26 C0 B0 80,

DL = –1

Округление:
REZ = 00 00 1B F9 27 ...

Знак результата:
80h XOR 00h = 80h (знак "–"),

DH = 80h
Порядок результата: PR = –4 + 3 – 1 = – 2



AL = 42h
REZF = C2 1B F9 27 (–0.000426837941 ( –0.000426837884)
26. Пример 11. Сложение чисел с плавающей точкой


Сложение чисел с плавающей точкой выполняется аналогично десятичной арифметике:


0.123*10–2 + 0.456*10–1 = 0.0123*10–1 + 0.456*10–1 = 0.4683*10–1
Отметим ряд особенностей операции сложения:

1. Выравнивание порядков. Выравнивается мантисса числа с меньшим порядком (сдвиг мантиссы "вправо").

2. Перевод мантиссы в отрицательное число сопровождается затиранием старшего байта числа:

старший байт C1 F5 A1 38 ( FF 0A 5E C8
                              мантисса     мантисса (отрицательное число)
поэтому старший байт нужно хранить отдельно.

3. Необходимо предусмотреть нормализацию "вправо" при переполнении результата, например:

                                               ( (сдвиг "вправо" на одну цифру)


F0 00 00 + F0 00 00 = 01 00 00 00

4. Перед нормализацией "влево" нужно проверить результат на ноль


5. Операции суммирования/вычитания имеют смысл при разнице порядков меньше 6:


11 11 11*161  (
11 11 11*161
P = 1 – (​–5) = 6


11 11 11*16–5 (
00 00 00 11 11 11*161 (за пределами мантиссы)

Шаги программной реализации:


1. Расчет разницы порядков P.


2. Выравнивание порядка числа с меньшим порядком.


3. Сохранение результирующего порядка PR, обнуление старших байт слагаемых.


4. Анализ знаков мантисс – перевод отрицательных мантисс в обратный код.


5. Суммирование мантисс.


6. Анализ знака результата – перевод отрицательного числа из обратного кода в положительное число.


7. Анализ нормализации "вправо" (переполнение), фиксация порядка нормализации РН.


8. Анализ нормализации "влево", фиксация РН.


9. Формирование старшего байта результата (знак мантиссы, знак порядка, порядок).
Пример: A1 = 80 1A 56 08, A2 = 41 F0 58 A1
1. PA1 = 0,
PA2 = –1,
P = 0 – (​–1) = 1

2. Выравнивание A2 = 41 0F 05 8A

3. PR = PA1 = 0,
A1 = 00 1A 56 08,

A2 = 00 0F 05 8A
4. Перевод в обратный код А1:

  1 00 00 00 00

 -  00 1A 56 08

    FF E5 A9 F8

5. Суммирование мантисс:

    FF E5 A9 F8

  + 00 0F 05 8A
    FF F4 AF 82
6. Анализ знака результата, перевод в положительное число:
  1 00 00 00 00

 -  FF F4 AF 82

    00 0B 50 7E
7. Нормализации вправо нет:
00 0B 50 7E
8. Нормализация "влево": 
00 0B 50 7E ( 00 B5 07 E0, PН = –1
9. PR = PR + PН = 0 – 1 = –1, 
REZA = C1 B5 07 E0 = –0,0441969633

10. Проверка в десятичной арифметике:

A1 + A2 = –0,1028752327 + 0,0586782731 = –0,0441969596 – совпадает до 6 десятичных цифр.
Программная реализация:


В программной реализации будем использовать индексные регистры SI, DI для обращения к слагаемым А1, А2. Причем SI – слагаемое с большим порядком, DI – с меньшим порядком (выравниваем слагаемое DI). Если разница порядков отрицательна: PA1 – PA2 < 0, то выполняется перестановка SI ( DI, чтобы SI указывал на слагаемое с большим порядком.


После приведения порядков слагаемое с большим порядком (SI) записывается в результат, а затем к нему прибавляется второе слагаемое (DI), например, A1 ( REZA, REZA = REZA + A2, где А1 – слагаемое с большим порядком. Поэтому регистр SI сначала указывает на слагаемое (например А1), а затем на результат (REZA).


Старший байт слагаемого SI будем хранить в регистре BH. Порядок нормализации – регистр DL, знак результата – DH. Сформированный старший байт результата – регистр BL.


Реализация сложения чисел с плавающей точкой более громоздкая чем умножение. В ней присутствуют три операции сдвига: выравнивание слагаемых, нормализация "вправо", нормализация "влево". Операции сдвигов двойных слов выполняются по одному биту командами SHR, SHL, RCR, RCL, например:
Выравнивание слагаемого А2 (DI) – сдвиг "вправо" на 4 бита * P
дв. слово в памяти        дв. слово при операциях сдвига

	1
	2
	3
	4
	(
	4
	3
	2
	1


                                           SHR    RCR
                                                  CF
При адресации к байтам (словам) двойного слова нужно использовать его представление в памяти, а при операциях сдвига – представление в регистрах. Сначала сдвигается вправо 3-й байт на один бит командой SHR, младший бит попадает в флаг переноса CF, затем сдвигается слово (1 и 2 байты) командой RCR, при этом бит из флага переноса попадает в старший бит сдвигаемого слова. Такие одиночные сдвиги выполняются 4*P раз:

        MOV AH,(4*dP)         ;в AH количество сдвигов
Sdvig:  SHR BYTE PTR [DI+2],1 ;сдвиг 3-го байта
        RCR WORD PTR [DI],1   ;сдвиг слова (1,2 байты)
        DEC AH                ;уменшить счетчик
        JNZ Sdvig

Сдвиги "влево", выполняются аналогично, но в обратной последовательности – сначала сдвигается влево слово (1 и 2 байты) командой SHL, затем сдвигается влево 3-й байт командой RCL.


Также значительную часть программной реализации занимают операции перевода положительного числа в отрицательное и наоборот. Эти операции выполняются методом инверсии числа и прибавления 1, например:
	A1 (SI)
	1
	2
	3
	4


        XOR DH,DH      ;DH = 0 (знак результата)
        MOV AH,[SI+3]  ;выборка в АН старшего байта
        MOV [SI+3],DH  ;обнуление старшего байта
        AND AH,80h     ;проверка знака
        JZ  Met        ;если знак "+" то уходим на Met
        MOV DX,[SI]    ;младшее слово А1 в DX (байты 1,2)
        NOT DX         ;инверсия DX
        ADD DX,1       ;DX+1
        MOV [SI],DX    ;возврат DX в младшое слово
        MOV DX,[SI+2]  ;старшее слово А1 в DX (байты 3,4)
        NOT DX         ;инверсия DX
        ADC DX,0       ;DX+CF(флаг переноса)

        MOV [SI+2],DX  ;возврат DX в старшее слово
Met:    ...
Полная реализация:
TITLE   ADDF

PUBLIC  A1,A2,REZA,ADDF

;-------------------------

DSEG    SEGMENT PARA PUBLIC

    A1      DD

    A2      DD

    REZA    DD       

DSEG    ENDS         

;-------------------------

CSEG    SEGMENT PARA PUBLIC

    ASSUME  CS:CSEG,DS:DSEG

;-------------------------

    ADDF    PROC    FAR

;-------------------------Расчет dP = |Pm1 - Pm2| -> AH

        MOV SI,offset A1

        MOV AH,[SI+3]     ;старший байт А1 -> AH

        MOV DI,offset A2

        MOV AL,[DI+3]     ;старший байт А2 -> AL

        MOV BH,AH

        AND AH,3Fh        ;порядок Pa1 -> AH

        AND BH,40h        ;знак порядка Pa1 -> BH
        JZ  Next          ;если знак Pa1 "+", то Next
        NEG AH            ;иначе меняем знак Pa1

Next:   MOV BL,AL

        AND AL,3Fh        ;порядок Pa2-> AL

        AND BL,40h        ;знак порядка Pa2-> BL
        JZ  SubP          ;если знак Pa2 "+", то SubP
        NEG AL            ;иначе меняем знак Pa2

SubP:   SUB AH,AL         ;dP = Pa1 - Pa2 -> AH

        JNS FillREZ       ;если AH > 0, то FillREZ

        NEG AH            ;иначе меняем знак AH
        MOV BX,DI         ;и выполняем перестановку

        MOV DI,SI         ;DI <-> SI

        MOV SI,BX

;-------------------------пересылка A1(SI) -> REZA

FillREZ:MOV BX,offset REZA

        MOV DX,[SI]

        MOV [BX],DX

        MOV DX,[SI+2]

        MOV [BX+2],DX

        MOV SI,BX         ; SI -> REZA

;-------------------------Проверка dP < 6

        CMP AH,6

        JL  AlignP        ;если AH >= 6 то завершить

        RET               ;процедуру

;-------------------------Выравнивание порядка

AlignP: MOV CL,2

        SHL AH,CL         ;АН = AH*4 (кол-во сдвигов)

        JZ  SetSign       ;если АН = 0 то SetSign
Sdvig:  SHR BYTE PTR [DI+2],1

        RCR WORD PTR [DI],1

        DEC AH

        JNZ Sdvig
;-------------------------Установка знаков мантисс A1, A2       

SetSign:XOR DH,DH         ;слагаемое SI
        MOV AH,[SI+3]     

        MOV [SI+3],DH    

        MOV BH,AH         ;сохранение старшего байта в BH
        AND AH,80h

        JZ  SignA2

        MOV DX,[SI]     

        NOT DX

        ADD DX,1

        MOV [SI],DX

        MOV DX,[SI+2]

        NOT DX

        ADC DX,0

        MOV [SI+2],DX

SignA2: XOR DH,DH         ;слагаемое DI

        MOV AH,[DI+3]    

        MOV [DI+3],DH

        AND AH,80h

        JZ  CalcREZ

        MOV DX,[DI]  

        NOT DX

        ADD DX,1

        MOV [DI],DX

        MOV DX,[DI+2]

        NOT DX

        ADC DX,0

        MOV [DI+2],DX

;-------------------------сложение REZA = REZA + A2

CalcREZ:MOV AX,[DI]     

        ADD [SI],AX     

        MOV AX,[DI+2]    

        ADC [SI+2],AX     

;-------------------------Проверка REZA = 0

        MOV AX,[SI]       

        MOV DX,[SI+2]

        OR  AX,DX

        JNZ ChkSign       ;если равен нулю

        RET               ;то завершение процедуры

;-------------------------Определение знака REZA -> DH
ChkSign:XOR DH,DH         ;DH = 0 (по умолчанию знак "+")

        MOV AL,[SI+3]       

        SHL AL,1

        JNC NormR         ;если знак "+" то NormR
        MOV AX,[SI]       ;иначе меняем знак REZA
        NOT AX

        ADD AX,1

        MOV [SI],AX

        MOV AX,[SI+2]

        NOT AX

        ADC AX,0

        MOV [SI+2],AX
        MOV DH,80h        ;и фиксируем его в DH = 80h
;-------------------------Нормализация вправо REZA
NormR:  XOR DL,DL         ;порядок нормализации Pн = 0

        MOV AX,[SI+2]

        AND AX,100h

        JZ  NormL         

        INC DL            ;Pн = +1

        MOV CL,4

SdvigR: SHR WORD PTR [SI+2],1

        RCR WORD PTR [SI],1

        DEC CL

        JNZ SdvigR

;-------------------------Нормализация влево REZA
NormL:  MOV AL,[SI+2]

        AND AL,0F0h

        JNZ CalcP         

        DEC DL            ;Pн = Pн –1

        MOV CL,4

SdvigL: SHL WORD PTR [SI],1

        RCL BYTE PTR [SI+2],1

        DEC CL

        JNZ SdvigL

        JMP NormL

;-------------------------Запись знака и порядка REZA
CalcP:  MOV BL,BH         ;старший байт слагаемого в BL
        AND BL,3Fh        ;выбрать порядок в BL
        AND BH,40h        ;знак порядка в BH
        JZ  SumP          ;если знак "+" то SumP
        NEG BL            ;иначе меняем знак

SumP:   ADD BL,DL         ;подсуммируем Pн

        JNS WrREZ         ;если рез. знак "+" то WrREZ
        NEG BL            ;иначе меняем знак

        OR  BL,40h        ;фиксируем знак порядка "-"

WrREZ:  OR  BL,DH         ;добавляем знак мантиссы

        MOV [SI+3],BL     ;запись BL в старший байт REZA
        RET

    ADDF    ENDP

CSEG    ENDS

END
27. Команды обработки строк

В данном контексте строка – это набор (массив) последовательных байт или слов. Для обработки строк используются пять основных операций:

Movs – пересылка байта или слова из адреса DS:[SI] по адресу ES:[DI] (только так). Полная запись выглядит так:

MOVS
byte ptr DS:[SI], byte ptr ES:[DI]
(
MOVSB;

MOVS
word ptr [DI], word ptr [SI]

(
MOVSW.

Запись операндов в команде MOVS нужна только для указания их типа. Очередность записи операндов не имеет значения. Сокращенная форма команды для пересылки байта или слова записывается MOVSB, MOVSW. После выполнения команды регистры SI, DI увеличиваются на 1 (пересылка байт) или на 2 (пересылка слов) если флаг DF = 0 (direct flag), если DF = 1, значения регистров уменьшаются на 1 или 2. 
Управление флагом DF выполняется командами: 
CLD (DF = 0),

SТD (DF = 1).

Cmps – сравнение байт или слов по адресам памяти DS:[SI] и ES:[DI]. Сокращенная форма команды: CMPSB, CMPSW. Регистры SI, DI изменяются на +1(+2) при DF = 0, или на –1(–2) при DF = 1.
Lods – загрузка аккумулятора AX(AL) из адреса памяти DS:[SI]. Сокращенная форма команды: LODSB, LODSW. Регистр SI изменяются автоматически.
STOS – запись аккумулятора AX(AL) в память по адресу ES:[DI]. Сокращенная форма команды: STOSB, STOSW. Регистр DI изменяются автоматически.
SCAS – сравнение содержимого аккумулятора AX(AL) и памяти по адресу ES:[DI]. Сокращенная форма команды: SCASB, SCASW. Регистр DI изменяются автоматически.
Для работы с набором байт или слов используется префикс повторения REP, например
REP
MOVSB

Количество циклов предварительно задается в регистре CX, команда REP автоматически выполняет декремент СХ и заканчивается при СХ = 0.

Для загрузки регистров SI, DI начальными адресами строк используется команда LEА, например запись:
LEA
SI, STRING
идентична


MOV
SI, offset STRING


Перед использованием команд пересылки необходимо инициализировать сегментный регистр ES, например:


MOV
AX, DSEG


MOV
DS, AX


MOV
ES, AX


Пример пересылки данных двух строк:
………………………………

STRING1
DB
20 DUP ('*')

STRING2
DB
20 DUP (?)

………………………………

CLD
MOV
CX, 20
LEA
SI, STRING1

LEA
DI, STRING2

REP
MOVSB

В данной реализации изменится содержимое строки STRING2, которая будет заполнена символом "*".

Команды сравнения CMPS (сравнение [SI], [DI]) и SCAS (сравнение AX, [DI]) используются с префиксами повторения REPE, REPZ или REPNE, REPNZ. REPE, REPZ выполняются пока флаг нуля ZF = 1 и CX ≠ 0 (до первого несовпадения). REPNE, REPNZ выполняются пока флаг нуля ZF = 0 и CX ≠ 0 (до первого совпадения).
Рассмотрим элементарные примеры.

CLD


LEA

SI, NAMЕ1

LODSW




CLD

LEA

DI, NAME

MOV

CX, 5

MOV

AX, 2020H

REP

STOSW


CLD

MOV

CX, 10

LEA

SI, NAME1

LEA

DI, NAME2

REPE

CMPSB

JNЕ

...

CLD

MOV

CX, 10

LEA

DI, NAME1

MOV

AL, ’M’

REPNE
SCASB

JNЕ

...
CLD

MOV

AL, ‘&’

MOV

CX, 15

LEA

DI, STRING

REPNE
SCASB

JNЕ
DEC

DI

MOV

[DI], 20h
FRAG
DB ‘***---‘

POLE

DB 15 DUP(?)

.........................................
CLD

MOV

CX, 15
LEA

SI, FRAG

LEA

DI, POLE

REP

MOVSB
28. Пример 12. Сортировка массива строк

Необходимо реализовать процедуру сортировки строк по возрастанию (по алфавиту). Строки задаются в формате: kS,'_ _ _ _', где kS – количество символов в строке (байт). Строки могут содержать разное количество символов. Также известно количество строк. Исходные данные имеют вид:

k
DW
...


;количество строк

Mas
DB
k1,'_ _ _ _',

;строка 1



k2,'_ _ _',

;строка 2



...




kK,'_ _ _ _ _'
;строка k

Сортировку реализуем по методу всплывающего пузырька: по очереди выбираются элементы массива с 2-го по k-й, текущий элемент сравнивается с предыдущим, если он меньше предыдущего, то выполняется перестановка элементов. Текущий элемент "всплывает" вверх по массиву пока он не окажется больше или равным предыдущему.

	0000

	
	
	
	
	



Перестановка заканчивается по условию: тек. элем. ( пред. элем., или когда достигается вершина массива.

Строки сравниваются посимвольно, до первого несовпадения. Сравниваются не сами символы а их числовые коды в таблице ASCII. Коды символов возрастают в соответствии с алфавитом символов, за счет этого можно сортировать строки в алфавитном порядке (могут быть исключения, например, символ украинское "і" имеет код меньше символа "а").


При разной длине строк сравниваются символы по размеру меньшей длины. Если получено равенство, то сравнение выполняется по длине строк, например:

'1234' > '123',
'abcd' > 'ab'


Большие и малые символы имеют разный код ASCII, поэтому для упрощения примем, что все символы строк записаны в одном регистре (либо все малые, либо большие символы).

Реализация сравнения строк:

AH


AL

k1 _ _ _ _ _ _ _ _ 
k2 _ _ _ _ _ _ _ _ _

DI


SI

Регистр SI указывает на текущую строку, регистр DI – на предыдущую строку. Длину предыдущей строки k1 запишем в регистр AH, длину текущей строки – в регистр AL.
MOV
AH,(k1)
;AH = k1

INC
DI          

LODSB

;AL = k2, Inc(SI)

XOR
CX,CX
MOV
CL,AL
;CX = k2

CMP
AL,AH 

JLE
Comp

;k2 <= k1 ?

MOV
CL,AH
;CX = k1

Comp:
REPE CMPSB
;посимвольное сравнение
JNE
Analiz
;если неравенство то на анализ
CMP
AL,AH
;а если равенство, то сравнение длин
Analiz:
JL
ChgStr
;если тек. < пред. то на перестановку строк
JMP
NextStr
;если нет, то переход к следующей строке

Реализациию перестановки строк будем выполнять циклически:

LODSB 1
     AL
10  STOSB
BX(
k1 _  _  _ k2 _  _  _  _  после k1+1 цикл. сдвигов ( k2 _  _  _  _ k1 _  _  _

    2  3  4  5  6  7  8  9   REP MOVSB
Цикл REP MOVSB выполняется k1+k2+1 раз. Циклический сдвиг повторяется k1+1 раз. Адрес на первый байт содержится в регистре BX. Цифрами 1,2,3, … 10 обозначена последовательность перемещения байт.
ChgStr:
XOR
CX,CX

;CX = 0
MOV
CL,AH

;CX = k1+1

INC
CL
ADD
AH,AL

;AH = k1+k2+1

INC
AH
Sdvig:
PUSH CX


;внешний цикл 
MOV
DI,BX

;адрес предыдущей строки
MOV
SI,DI
LODSB


;первый символ ( AL
MOV
CL,AH
REP
MOVSB

;циклический сдвиг символов
STOSB


;AL ( в последний символ

POP CX
LOOP  Sdvig

;внешний цикл

Программная реализация:
title prim9

stek    segment stack

    db  64  dup(?)

stek    ends 

dseg    segment

    k   dw  5

    Mas db  3,'one',3,'two',5,'three',4,'four',4,'five'  

dseg    ends

cseg    segment

    assume ds:dseg, cs:cseg, ss:stek

;загрузка сегментов ds,es

Start:  mov     ax,dseg

        mov     ds,ax

        mov     es,ax

        cld                 ;флаг df = 0 (di+,si+)

;организация основного цикла по массиву строк Mas
        xor     cx,cx 

        push    cx

NextStr:pop     cx

        inc     cx

        cmp     cx,k

        je      Finish

        push    cx

        mov     dx,cx   ;сохранение позиции для всплывания 

;выборка текущей строки, si - текущая, di - предыдущая

NextUp: mov     si,offset Mas

        xor     ax,ax

SelStr: mov     di,si

        lodsb               ;al = k1, inc(si)     

        add     si,ax

        loop    SelStr

;сохранение адреса предыдущей строки в bx

        mov     bx,di

;сравнение текущей и предыдущей строк

        mov     ah,al       ;ah = k1

        inc     di          

        lodsb               ;al = k2, inc(si)

        xor     cx,cx

        mov     cl,al       ;cx = k2

        cmp     al,ah       

        jle     Comp        ;k2 <= k1 ?

        mov     cl,ah       ;cx = k1

Comp:   repe    cmpsb       ;посимвольное сравнение

        jne     Analiz

        cmp     al,ah

Analiz: jl      ChgStr

        jmp     NextStr

;перестановка строк

ChgStr: xor     cx,cx

        mov     cl,ah       ;cx = k1+1

        inc     cl

        add     ah,al       ;ah = k1+k2+1

        inc     ah          

Sdvig:  push    cx

        mov     di,bx       ;адрес предыдущей строки
        mov     si,di

        lodsb               ;первый символ --> al

        mov     cl,ah

        rep     movsb       ;циклический сдвиг символов

        stosb               ;al --> в последний символ

        pop     cx

        loop    Sdvig

;всплываем выше или выбираем следующую строку

        dec     dx

        mov     cx,dx

        jnz     NextUp

        jmp     NextStr

Finish: mov     ax,4c00h    ;завершение программы
        int     21h

cseg ends

end Start
Комментарий:

1. После загрузки сегментов DS, ES выполняется команда CLD для сброса флага DF = 0, чтобы все команды обработки строк (LODSB, CMPSB, MOVSB, STOSB) сопровождались увеличением регистров SI, DI.

2. По метке NextStr выбирается новый текущий элемент, по метке NextUp выбирается текущий элемент для "всплывания".

3. Текущая строка выбирается по счетчику CX: CX = 1 (2-я строка), CX = 2 (3-я строка) и т.д. Процесс выборки одинаков для новой строки и текущей строки после "всплытия". После выборки регистр SI указывает на текущую строку, а DI – на предыдущую.
4. Регистр DX отвечает за контроль достижения вершины при "всплывании" (если DX = 0, то достигнута вершина).

5. В результате сортировки массив Mas будет иметь вид:

4five4four3one5three3two
29. Алгоритмы ускорения доступа к данным

Задача поиска информации является одной из наиболее распространенных среди промышленного использования компьютерной техники. Работа с справочниками, словарями, каталогами оборудования, информационные системы технических и финансовых сфер, компоненты системного математического обеспечения, системы паролей, модели функционального программного обеспечения и т.д. требуют поиска информации по некоторым ключам и требуют минимизации временных затрат на этот процесс. Для формализации алгоритмов ускорения будем рассматривать упрощенные информационные структуры, состоящие из записей фиксированной длины. Каждая запись имеет собственный ключ.
структура DAN

	
	Key1
	DAN1
	+0*K

	Key
	Key2
	DAN2
	+1*K

	
	Key3
	DAN3
	+2*K


где
N – 
количество записей

K – 
размер записи
Key –
поисковый ключ требуемой записи

Последовательный поиск
Пусть ключ (Key) является целым числом и занимает слово, т.е. может принимать значения 0000÷FFFF (в шестнадцатеричном виде), однако это не значит что реальное количество записей равно предельному значению ключа. Это может быть массив в диапазоне 1000÷10000 элементов – объем после которого можно задумываться об ускорении.

      MOV
CX, N


;организация цикла поиска

      MOV
SI, offset DAN
;выборка первой записи
      MOV
AX, Key


;начальная загрузка

Find: CMP
DAN[SI], AX

      JE

...



;элемент найден

      ADD
SI, K

      LOOP
Find
Последовательный поиск является самым простым по реализации, однако, и наиболее не эффективным. Очевидно, что эффективными методами поиска являются те методы, которые существенно уменьшают число сравнений.
Метод половинного деления


Для реализации метода необходимо упорядочить массив ключей по возрастанию

1


i2
i3
i4
i1

i = (min+ max)/2
        N
DAN

(

          >
>
=       <


     K

        (

min

     min=i2  min=i3
      max = i1




        max

В начальном состоянии min = 1, max = N. На очередном шаге поиска вычисляется индекс i = (min + max)/2 (целочисленное деление, например, 3/2 = 2, 4/2 = 2, 5/2 = 3 и т.д.), если Key < DAN[i], то max = i, если Key > DAN[i], то min = i, затем повторяется вычисление индекса i. Шаги повторяются до выполнения условия Key = DAN[i]. На каждом шаге область поиска сокращается вдвое. Максимум шагов сравнения m определяется по выражению 2m > N, например, для N = 10 ( m = 4, для N = 1000 ( m = 10, для N = 100000 ( m = 17.

В алгоритме поиска необходимо предусмотреть ситуацию, когда значение Key отсутствует в структуре DAN. В этом случае выход осуществляется по условию min = max. Алгоритм поиска методом половинного деления показан ниже:




min = 1, max = N




|



i = (min + max)/2




|



f = Key – DAN[i]





|



f = 0 ?  
(да)




Выход




| (нет)



min = max ?
 (да)




Выход




| (нет)



min = max–1 ? (нет)





| (да)



f > 0 ?
(да)




| (нет)



max = min



min = max



f > 0 ?
(да)





| (нет)




max = i



min = i

После выхода из алгоритма поиска индекс i указывает на искомую запись. Если f ( 0, то запись не найдена.
Пример:
      i
=
1
2
3
4
5
6
7
8
9
10
11
12
DAN
=
1
2
5
7
9
11
15
20
24
30
41
55
Key = 20,
min = 1,
max = 12
i = (1 + 12)/2 = 6

20 > 11
min = 6
i = (6 + 12)/2 = 9

20 < 24
max = 9

i = (6 + 9)/2 = 7

20 > 15
min = 7

i = (7+ 9)/2 = 8

20 = 20
выход

Key = 17

i = (7+ 9)/2 = 8

17 < 20
max = 8
i = (7+ 8)/2 = 7

17 > 15
min = max – 1 ( min = max = 8

i = (8+ 8)/2 = 8

min = max
выход
Если ключ не найден, то после выхода индекс i указывает на то место, где должен быть Key. Например, Key = 17 должен быть по индексу i = 8 (при условии, что остальные ключи в массиве DAN сдвинуться вправо).
30. Пример 13. Реализация поиска методом половинного деления
DSEG

SEGMENT

N
DW ...       ;количество записей в структуре DAN

K
DW ...       ;размер одной записи

Key
DW ...       ;искомый ключ

DAN
DW ...       ;структура DAN
DSEG

ENDS
CSEG

SEGMENT

ASSUME
DS:DSEG, CS:CSEG
Start:
MOV AX, DSEG ;загрузка сегмента данных


MOV DS, AX


MOV BX,1     ;min = 1


MOV DX,N     ;max = N

Next:
MOV AX,DX    ;i = (min + max)/2 ( AX


ADD AX,BX



SHR AX,1



MOV DI,AX    ;сохранение i в DI



DEC AX       ;i-1 ( AX


MUL K       ;(i-1)*K(AX расчет адреса записи DAN


MOV SI,AX



MOV AX,DAN[SI] ;выборка очередного ключа в AX


MOV CX,Key   


SUB CX,AX    ;f = Key – DAN[SI] ( CX


JZ Finish    ;если ноль (f = 0) то на Finish


CMP BX,DX    ;min = max ?



JZ Finish    ;если да то Finish


MOV AX,DX



SUB AX,BX    ;AX = max - min



CMP AX,1     ;max – min = 1?



JNZ MinMax   ;если нет то MinMax
CMP CX,0     ;если да то проверяем


JG Min       ;f > 0 ?



MOV DX,BX    ;если нет то max = min


JMP Next     ;следующий шаг сравнения
Min:

MOV BX,DX    ;если да то min = max


JMP Next

MinMax:
CMP CX,0     ;проверяем


JG MinIdx    ;f > 0 ?



MOV DX,DI    ;если нет то max = i


JMP Next     ;следующий шаг сравнения
MinIdx:
MOV BX,DI    ;если да то min = i


JMP Next

Finish:
MOV AX,4C00h ;функция завершения программы AH=4C


INT 21h      ;и выход без ошибок AL=0
CSEG

ENDS

END
Start
Комментарий:

1. В регистрах BX, DX хранятся значения min, max.

2. Текущий индекс i (1...N) хранится в регистре DI, смещение по структуре DAN рассчитывается как SI = (i-1)*K.

3. Значение флага f = Key – DAN[SI] определяется регистром CX.


4. Метод существенно сокращает время поиска данных по ключу, однако, при добавлении новой записи требует упорядочения структуры DAN. Время упорядочения может быть существенным, особенно в ситуациях частого добавления записей.
31. Хеширование

В этом методе и в основе алгоритмов хеширования лежит хеш-функция, которая преобразует заданный ключ в непосредственный адрес информационной компоненты или в адрес хеш-таблицы, которая разворачивает методами ссылок переходы в информационные компоненты. Идеальной хеш-функцией является функция которая для любых двух неодинаковых ключей дает неодинаковые адреса К1 ≠ К2 => h(К1) ≠ h(К2). Естественно, размер адреса определяется объемом памяти выделенной от хеш-таблицы. С одной стороны хеш-функция должна быть достаточно простая и быстродействующая, с другой стороны она должна обеспечивать равномерную адресацию в поле хеш-таблицы и как можно меньше создавать коллизий. Коллизия – это ситуация, когда два ключа дают один и тот же адрес К1 ≠ К2 => h(К1) = h(К2). Эта ситуация связана с переходом от большего множества к меньшему множеству. Хеш-функция необратима (из хеш-адреса нельзя однозначно получить исходный ключ).

Для числовых ключей простейшим вариантом хеш-функции является функция MOD (остаток от деления), например, функция h(Key) = Key MOD 1000 (ключ делится на 1000, а остаток равен хеш-адресу). Разрешение коллизий может быть выполнено на основе списочных структур. В них поиск может осуществляться последовательно.
Пример:

Пусть имеем следующий состав ключей:

75
66
42
192
91
40
49
87
67
16
417
130
372
227

В качестве хеш-функции выберем 

h(Кey) = Key MOD 10 
– остаток от деления на 10. Максимальное количество полей хеш-таблицы равно 10 (0…9).
	хеш-

адрес
	хеш-

таблица
	
	
	
	
	
	
	
	

	0
	
	→
	40
	→
	130
	
	
	
	

	1
	
	→
	91
	
	
	
	
	
	

	2
	
	→
	42
	→
	192
	→
	372
	
	

	3
	
	
	
	
	
	
	
	
	

	4
	
	
	
	
	
	
	
	
	

	5
	
	→
	75
	
	
	
	
	
	

	6
	
	→
	66
	→
	16
	
	
	
	

	7
	
	→
	87
	→
	67
	→
	417
	→
	227

	8
	
	
	
	
	
	
	
	
	

	9
	
	→
	49
	
	
	
	
	
	



Алгоритм поиска здесь реализуется путем преобразования ключа в хеш-адрес, по хеш-адресу входят в хеш-таблицу, а дальше по списочной структуре методом последовательного поиска. Алгоритм не требует упорядочения. Списочные структуры хорошо реализуются в динамической памяти, обеспечиваются простейшие варианты добавления новых записей и их удаления.

32. Реализация функций-обработчиков прерываний

В составе вектора прерываний (номера 0…255) содержатся внутренние прерывания (например, деление на ноль, недопустимый код операции, пошаговый режим), аппаратные прерывания (например, системный таймер, клавиатура, порты, работа с дисками) и программные прерывания (BIOS, DOS, пользовательские).


Внутренние и аппаратные прерывания называют асинхронными, так как они могут возникнуть в произвольный момент выполнения программы. Программные прерывания называют синхронными, так как они являются набором функций, и всегда вызываются в заданной точке программы. В векторе прерываний зарезервированы ряд номеров (например, 60h...66h) для возможности программирования пользовательских прерываний.


Напомним, что вызов прерывания выполняется командой INT, например INT 21h (диспетчер функций DOS). Команда INT обращается к вектору прерываний, по номеру прерывания, умноженному на 4, выбирает адрес функции-обработчика прерываний и передает ей управление. Адреса функций-обработчиков прерываний находятся в самом начале оперативной памяти с адреса 0 до 1020.
                                                                                                                 Функция-
                                                       Адрес функции                                обработчик
               Вектор прерываний      обработчика      Программа            прерывания
Память

              0h      4h      8h      Ch …60h*4                 ... INT 60h ...

В векторе прерываний можно установить адрес собственной функции-обработчика прерывания с помощью функции AH = 25h диспетчера DOS (INT 21h). Полный адрес собственной функции должен быть предварительно записан регистрах DS:DX. Номер прерывания записывается в регистр AL. Например, подстановка функции my_int для пользовательского прерывания 60h имеет вид:
my_int
PROC FAR ;собственный обработчик прерывания 60h


...



IRET     ;завершение прерывания командой IRET
my_int
ENDP


...



PUSH DS  ;сохранение регистра DS


PUSH CS  ;пересылка регистра CS


POP DS   ;в регистр DS


MOV DX,offset my_int ;смещение my_int в DX


MOV AX,2560h ;номер функции AH = 25h, 


INT 21h      ;номер прерывания AL = 60h


POP DS   ;восстановление регистра DS


...



INT 60h  ;вызов прерывания 60h

Собственный обработчик прерывания должен заканчиваться командой IRET. Для формирования полного адреса функции (сегмент:смещение) выполняется пересылка регистра кода CS в регистр данных DS. После выполнения подстановки любой вызов INT 60h будет передавать управление функции my_int.

Для пользовательских прерываний вызов INT 60h эквивалентен вызову CALL my_int. Преимущества вызова INT появляются в случае резидентной загрузки функции my_int. В этом случае к ней могут обращатся любые программы посредством вызова INT 60h.

С помощью приведенных операций можно выполнить подстановку обработчика и для системных прерываний. В этом случае в начале программы сохраняется адрес системного обработчика, затем выполняется подстановка адреса собственного обработчика, а в конце программы восстанавливается адрес системного обработчика. Чтение адреса системного обработчика выполняется функцией DOS AH = 35h, в AL должен быть записан номер прерывания. Полный адрес функции-обработчика читается в регистры ES:BX.


old_int  DD      ;объявление переменной old_int


...



MOV AX,351Ch     ;номер прерывания AL = 1Ch


INT 21h          ;адрес обработчика ( ES:BX


MOV old_int,BX   ;сохранение адреса обработчика


MOV old_int+2,ES ;в переменную old_int

Обработчик прерываний 1Ch служит для обработки прерываний от системного таймера и состоит из одной команды IRET. Он вызывается 18.2 раза в секунду.

При корректной подстановке системных обработчиков прерываний внутри пользовательского обработчика вызывается системный обработчик, например:


my_int PROC FAR




PUSHF




CALL DWORD PTR old_int




...




IRET



my_int ENDP


Команда PUSHF сохраняет в стек регистр флагов для команды IRET системного обработчика old_int.

В случае с прерыванием 1Ch в этом нет необходимости, так как исходный обработчик состоит из одной команды IRET. А если использовать подстановку аппаратного прерывания 08h от системного таймера, то вызов исходного обработчика внутри пользовательского обязателен. Иначе на время работы программы часы компьютера изменяться не будут.


Обработчики аппаратных прерываний должны заканчиваться командой EOI (End Of Interrupt) в контроллер прерываний:



...



MOV AL, 20h



OUT 20h, AL



IRET

Команда OUT выводит в порт 20h (контроллер прерываний) код AL=20h. При этом снимается блокировка прерываний текущего и низших уровней. Ниже приведена упрощенная схема контроллера прерываний.

	
	IRQ
	Контроллер

прерываний
	
	Схема анализа 

приоритетов
	

	Таймер
	0
	08h
	
	
	
	0
	
	

	Клавиатура

	1
	09h
	Базовый 
	
	
	0
	
	

	Ведом. контр.
	2
	0Ah
	вектор
	
	
	0
	
	

	COM2
	3
	.
	08h
	
	
	0
	
	

	1    (        Мышь
	4
	.
	   1     (
	
	
	1   0
	   (
	обработчик

	LPT2
	5
	.
	Порты
	
	
	1   0
	
	прерывания

	Гибк. диск
	6
	.
	20h, 21h
	
	
	1   0
	
	

	1    (     Принтер
	7
	0Fh
	   1     (
	
	
	1   0
	
	

	
	
	
	
	
	
	
	
	

	
	
	
	
	порт 20h
	
	

	
	
	
	
	
	
	EOI
	
	



Контроллер прерываний содержит восемь входных линий. Каждая линия имеет соединение с конкретным устройством компьютера. Эти линии нумеруются IRQ0, IRQ1, IRQ2 и т.д., где IRQ – interrupt request (запрос прерывания). Для каждого устройства вектор (номер) прерывания формируется  в виде суммы базового вектора 08h и номера линии, например, для линии IRQ0 вектор равен 08h, для линии IRQ1 – 09h и т.д. Восьми линий мало для обработки запросов от всех устройств, поэтому на компьютере имеется ведомый контроллер прерываний, подключаемый к ведущему по линии IRQ2. Приоритеты прерываний располагаются по спаданию от IRQ0 до IRQ7, т.е. наивысший приоритет имеет линия IRQ0 от системного таймера.

Внутри контроллера прерываний содержаться ряд регистров, через которые проходят запросы от устройств, для вызова соответствующего обработчика прерывания устройства (регистр запросов, регистр маски, схема приоритетов, регистр обслуживаемых запросов). Обращение к этим регистрам выполняется через порты 20h, 21h. Сигнал запроса прерывания сначала поступает на регистр запросов и сохраняется там до попадания в регистр обслуживаемых запросов. От регистра запросов сигнал проходит через регистр маски – в этом регистре можно установить запрет прерываний по отдельным линиям. После этого сигнал поступает на схему анализа приоритетов, где из нескольких запросов выбирается запрос с высшим уровнем приоритета, и устанавливается запрет для данного и нижних уровней запросов. Затем сигнал поступает в регистр обслуживаемых запросов, обнуляется в регистре запросов, и выполняется запуск обработчика прерываний для данного запроса от устройства. Таким образом выполнение обработчика прерывания не может быть прервано запросом устройств данного или нижних уровней. После окончания работы обработчика в контроллер прерываний посылается сигнал EOI для сброса запрета прерываний текущего и нижних уровней в схеме анализа приоритетов.
33. Пример 14. Обработка прерываний от таймера


Современные компьютеры оснащаются двумя подсистемами таймеров, параллельно отсчитывающими текущее время. Один таймер расположен в микросхеме с низким потреблением энергии – КМОП-микросхеме (комплементарная логика, металл окисел полупроводник, комплементарная - взаимно-дополняющая логика на полупроводниках с n-p и p-n переходами). Эта микросхема при выключении питания компьютера продолжает работать, получая энергию от встроенного в компьютер аккумулятора (батарейки). Этот таймер обычно называют часами реального времени. Помимо кварцевого генератора и систем управления он содержит внутреннюю память, в которой хранятся и наращиваются значения текущей даты и времени. Обновление времени осуществляется каждую секунду, причем эта операция выполняется на аппаратном уровне, не затрагивая ни процессор, ни основную оперативную память.

Другой таймер работает, как и все остальные узлы компьютера, только когда компьютер включен. Его называют системный таймер. Он вырабатывает сигналы с частотой примерно 18.206 Гц, вызывающие аппаратные прерывания уровня 0 (IRQ0 – вектор 08h). Обработчик этих прерываний, входящий в систему BIOS, с каждым прерыванием увеличивает на 1 содержимое двухсловной ячейки с адресом 40h:6Ch, расположенной в области данных BIOS. В процессе начальной загрузки компьютера программа BIOS считывает показания часов реального времени (часы, минуты, секунды) и, преобразовав их в число секунд, истекших от начала текущих суток, умножает на 18.206, чтобы получить текущее время, выраженное в числе тактов системного таймера. Эта величина заносится в ячейку с адресом 40h:6Ch. Выполняемый в дальнейшем инкремент этой ячейки поддерживает в ней правильное текущее время, пока компьютер включен. Именно из этой ячейки черпается информация о текущем времени программами DOS: функциями 2Ch или командой TIME.

Подсистема таймера работает независимо от процессора (параллельно с ним) от собственного генератора, вырабатывающего сигналы с частотой 1.19318 МГц. Таймер имеет счетчик, пересчитывающий сигналы от генератора, и регистр-фиксатор, в который программно заносится число, определяющее коэффициент пересчета счетчика. При включении компьютера в регистр-фиксатор заносится максимально возможное число FFFFh (65535) в результате чего сигналы на выходе системного таймера имеют частоту 1193180 Гц / 65535 = 18.206 Гц. Эти сигналы вызывают прерывания с вектором 08h, осуществляющий отсчет текущего времени.


В качестве примера реализуем функцию обработки телеизмерений. Измерения поступают с разной периодичностью:
· измерение тока – 1 сек

· измерение напряжения – 2 сек

· измерение мощности – 5 сек

Для отслеживания времени будем встраиваться в обработчик аппаратного прерывания от системного таймера 08h. Поскольку частота на выходе таймера равна 18.2 Гц, то 1 сек ( 18 тактов, 2 сек ( 36 тактов, 5 сек ( 91 такт.
Процедура обработки измерений будет выводить на экран соответствующий текст, например

msg1 DB 'Замер тока ...      ',0Ah,0Dh,'$'
                 ;строка заканчивается символом '$'

...

MOV DX,offset msg1 ;адрес строки в DS:DX
MOV AH,09h         ;номер функции 09h
INT 21h            ;вывод строки на экран

Параллельно будем выводить время от часов реального времени:


MOV AH,02h ;номер функции для чтения времени


INT 1Ah    ;CH-часы, CL-минуты, DH-секунды

           ;в формате BCD



           ;если CF=1 то часы не работают

Код BCD – двоично-десятичный код, в каждой тетраде (шестнадцатиричной цифре) могут быть цифры 0…9. Например, содержимое CH = 12h, CL = 45h, DH = 38h означает 12 часов, 45 минут, 38 секунд.

Для того, чтобы перевести содержимое регистров в строку для вывода на экран необходимо каждую тетраду записать в отдельный байт и прибавить 30h, чтобы получить коды символов '0', '1', ...'9'.
Программная реализация:

STK     SEGMENT STACK

    DB  64 DUP(?)

STK     ENDS

DSEG    SEGMENT

    msg1    DB  'Замер тока ...      ',0Ah,0Dh,'$'

    msg2    DB  'Замер напряжения ...',0Ah,0Dh,'$'

    msg3    DB  'Замер мощности ...  ',0Ah,0Dh,'$'

    shift   DB  23

DSEG    ENDS

CSEG    SEGMENT

    ASSUME  CS:CSEG,DS:DSEG,SS:STK

    timer_int   PROC    FAR

        PUSHF

        CALL dword ptr CS:int_08h

        PUSH AX

        PUSH BX

        PUSH CX

        PUSH SI

        XOR SI,SI

        MOV CX,CS:count

prc:    CALL analiz

        INC SI

        LOOP prc

        MOV AL,20h  ;сигнал EOI
        OUT 20h,AL  ;в контроллер прерываний
        POP SI

        POP CX

        POP BX

        POP AX

        IRET

        analiz  PROC

            MOV BX,SI

            SHL BX,1

            INC CS:timer[BX]
            MOV AX,CS:delay[BX]

            CMP AX,CS:timer[BX]

            JG  fin

            MOV CS:flags[SI],1

            MOV CS:timer[BX],0

fin:        RET

        analiz  ENDP

        int_08h DD

        delay   DW  18,36,91

        timer   DW  0,0,0

        flags   DB  0,0,0

        count   DW  3

    timer_int   ENDP

    out_info    PROC   

        MOV CS:flags[SI],0

        CALL real_time

        MOV AX,SI

        MUL shift

        MOV DX,BX

        ADD DX,AX

        MOV AH,09h

        INT 21h

        RET
    out_info    ENDP

    real_time    PROC 

        MOV AH,02h

        INT 1Ah

;выборка минут (освобождение CL)

        MOV AL,CL

        AND AL,0Fh

        ADD AL,30h

        MOV CS:rtm+4,AL

        MOV AL,CL

        AND AL,0F0h

        MOV CL,4

        SHR AL,CL

        ADD AL,30h

        MOV CS:rtm+3,AL

;выборка часов CH

        MOV AL,CH

        AND AL,0Fh

        ADD AL,30h

        MOV CS:rtm+1,AL

        MOV AL,CH

        AND AL,0F0h

        MOV CL,4

        SHR AL,CL

        ADD AL,30h

        MOV CS:rtm,AL

;выборка секунд DH

        MOV AL,DH

        AND AL,0Fh

        ADD AL,30h

        MOV CS:rtm+7,AL

        MOV AL,DH

        AND AL,0F0h

        SHR AL,CL

        ADD AL,30h

        MOV CS:rtm+6,AL

        PUSH DS

        PUSH CS

        POP DS

        LEA DX,CS:rtm

        MOV AH,09h

        INT 21h

        POP DS

        RET

        rtm     DB  0,0,':',0,0,':',0,0,'  $'

    real_time    ENDP

    main    PROC    FAR

;Загрузка регистра DS

        MOV AX,DSEG

        MOV DS,AX

;Зануление timer, flags

        PUSH CS

        POP ES

        XOR AX,AX

        MOV CX,CS:count

        LEA DI,CS:timer

        REP STOSW

        MOV CX,CS:count

        LEA DI,CS:flags

        REP STOSB

;Сохранение вектора 08h

        MOV AX,3508h

        INT 21h

        MOV word ptr CS:int_08h,BX

        MOV word ptr CS:int_08h+2,ES

;Подстановка собственного вектора обработчика

        PUSH DS

        MOV DX,offset timer_int

        MOV AX,CS

        MOV DS,AX

        MOV AX,2508h

        INT 21h

        POP DS

;функциональная часть программы

rpt:    LEA BX,msg1

        MOV CX,CS:count

        XOR SI,SI

inf:    CMP CS:flags[SI],0

        JZ  next

        PUSH CX

        CALL out_info

        POP CX

next:   INC SI

        LOOP inf

        MOV AH,06h

        MOV DL,0FFh

        INT 21h

        JZ  rpt

        CMP AL,1Bh

        JNE rpt

;Восстановление вектора 08h

        MOV AX,2508h

        LDS DX,int_08h

        INT 21h

;Завершение программы   

        MOV AX,4C00h

        INT 21h 

    main    ENDP

CSEG    ENDS

END     main
Комментарий:

1. В начале собственного обработчика timer_int вызывается системный обработчик.


2. Все используемые регистры сохраняются в стеке.


3. В процедуре analiz увеличиваются на 1 содержимое полей timer и сравнивается с задержкой в поле delay. При достижении равенства устанавливается 1 в соответствующее поле flags, а timer сбрасывается в ноль.

4. Переменные, используемые в процедуре timer_int хранятся непосредственно в теле процедуры, поэтому обращение к ним осуществляется через сегмент кода CS. Как правило, все переменные содержат в сегменте данных, на который указывает регистр DS. Однако в процедуре real_time для вывода строки регистр DS временно указывает на сегмент кода, и если прерывание от таймера возникнет именно в этот момент, то обращение к данным через регистр DS приведет к ошибке. Именно поэтому в обработчиках прерываний, как правило, переменные объявляют непосредственно в теле процедуры.


5. В процедуре timer_int перед восстановлением регистров из стека в контроллер прерываний посылается сигнал EOI, разрешающий выполнение прерываний данного и низших уровней.


6. Процедура out_info обнуляет flags для соответствующего измерения (индекс SI), вызывает процедуру real_time для вывода текущего времени на экран и выводит строку для соответствующего измерения. Для упрощения все строки имеют одинаковую длину, записанную в поле shift = 23 символа.

7. Процедура real_time реализует чтение часов реального времени и выдачи их на экран.


8. Процедура main в начале содержит подстановку собственного вектора 08h, а в конце – восстановление системного вектора. В функциональной части выполняется анализ flags, и если 1, то вызывается процедура обработки измерения out_info. Функциональная часть зациклена в бесконечный цикл, выход из которого осуществляется по клавише Esc.

9. После запуска программы на экране должны последовательно отображаться строки, например:


14:15:01 Замер тока …


14:15:02 Замер тока …
14:15:02 Замер напряжения …

14:15:03 Замер тока …


14:15:04 Замер тока …
14:15:04 Замер напряжения …

14:15:05 Замер тока …
14:15:05 Замер мощности …

10. Примерно через каждые полторы минуты появится дублирование строки:

14:16:30 Замер тока …


14:16:30 Замер тока …
что связано с погрешностью: 1 сек = 18.2 такта, т.е. за 1 сек погрешность составляет 0.2 такта. Погрешность составит 18 тактов за 18/0.2 = 90 сек. Для разных временных интервалов будут разные погрешности, например, 5 сек = 91.03 такта:


5 сек 

– погрешность 0.03 такта


15 167 сек
– погрешность 91 такт.

34. Система ввода-вывода

Логическую схему компьютера можно представить в виде системной шины (магистрали) к которой подключается сам процессор и все устройства компьютера.



Процессор связан с системной шиной практически всеми своими линиями из которых можно выделить две группы: линии адреса (шина адреса), линии данных (шина данных) и линия M/IO (Memory / In-Out). При чтении данных из памяти, процессор выставляет на шине адреса требуемый адрес памяти. Устройство памяти, расшифровав этот адрес, устанавливает на шине данных значение соответствующей ячейки памяти, после чего процессор считывает это значение в соответствующий регистр (MOV AX, mem). При записи в память, процессор выставляет адрес и данные на системной шине, а устройство памяти записывает эти данные по указанному адресу (MOV mem, AX).

Процедуры обмена данными с другими устройствами аналогичны памяти. За каждым устройством закреплена группа адресов, на которые они должны откликаться. Обнаружив свой адрес на магистрали устройство выполняет чтение или запись данных. Адреса устройств перекрываются адресами памяти. Их разграничение выполняется линией M/IO, если на линии 1 (memory) то это адрес памяти, если 0 (in-out), то адрес устройства ввода-вывода.

В языке Assembler обращение к любым устройствам ввода-вывода выполняется всего двумя командами: IN – чтение данных с устройства, OUT – запись данных на устройство. Адрес устройства должен быть записан в регистре DX, данное – в регистре AX или AL, например:



MOV DX, адрес устройства



IN  AX,DX  ;чтение данных с устройства


OUT DX,AL  ;запись данных в устройство

Устройства принято называть портами. Адресное пространство портов составляет 64 кбайт (максимум DX = FFFFh). Реально из этого пространства используется лишь очень малая часть.
35. Программирование последовательных интерфейсов (COM порты)

Последовательный интерфейс предназначен для передачи данных и использует одну сигнальную линию в одном направлении. Информационные биты передаются друг за другом последовательно.

При отсутствии данных в линии действует сигнал маркера (логическая 1). Передача данных начинается с посылки старт-бита (логический 0). После этого передаются биты данных количеством от 5 до 8, также называемые символом. Дальше может следовать бит контроля четности. Этот бит всегда выбирается так, чтобы общее число 1 битов данных и бита четности всегда было четным (или нечетным). Кадр заканчивается стоп-битом (логическая 1).

Сигналы в линии могут иметь различное представление. При передаче на небольшое расстояние логической 1 соответствует –15 В, логическому 0 – +15 В (стандарт RS-232). При расстояниях до 1.5 км используют токовую петлю – импульсы постоянного тока 20 или 40 мА. А при передаче на неограниченное расстояние по телефонным линиям уровни напряжений преобразуют в пачки синусоидальных сигналов: 1 – высокая частота, 0 – низкая частота.


Обычно компьютер имеет два встроенных последовательных интерфейса – COM1, COM2. Как правило, COM1 используется для подключения мыши, COM2 – для модема или связи между компьютерами.


Программирование последовательной передачи включает два этапа: инициализацию интерфейса и передачу-прием данных. На этапе инициализации необходимо задать скорость передачи и характеристики кадра: число бит данных, число стоп-битов, контроль четности и др. Эти настройки должны совпадать на передающей и принимающей стороне. Управление передачей можно выполнить функциями прерывания 14h BIOS или на физическом уровне, обращаясь непосредственно к регистрам последовательного порта. Будем использовать второй вариант.


У контроллера последовательного интерфейса имеется 8 регистров, занимающих смежные адреса в пространстве адресов устройств ввода-вывода. Регистр 0 служит для приема-передачи. Со стороны передатчика в него нужно поместить передаваемый байт, со стороны приемника из него читается принятый байт. Регистры 0 и 1 выполняют двоякие функции в зависимости от бита 7 регистра 3 (бит управления DLAB). Когда этот бит равен 0 (нормальное рабочее состояние) регистр 0 является регистром приемо-передатчика, а регистр 1 – задает маску прерываний. Когда этот бит равен 1, то регистры 0 и 1 применяются для чтения/записи старшего и младшего байтов делителя, определяющего скорость обмена. Значение делителя определяется по формуле:



K = 115 200 / V, 

где V – скорость передачи, бит/сек.
Входная частота синхронизации 1.8432 МГц делится на К и получается 16-кратная частота передачи.


Базовый адрес COM портов определяется путем чтения данных BIOS. Стандартным базовым адресом порта COM1 является 3F8h, COM2 – 2F8h. Адреса регистров СOM порта рассчитываются как сумма базового адреса и номера регистра (0…7).

Ниже приведено описание регистров COM порта:

	№

регистра
	DLAB (бит

управления)
	Описание

	0
	1

0
	младший байт делителя скорости обмена
регистр приема и передачи данных

	1
	1
0
	старший байт делителя скорости обмена
регистр маски разрешения прерываний

	2
	
	регистр идентификации прерываний

	3
	
	регистр управления линией

	4
	
	регистр управления модемом

	5
	
	регистр состояния линии

	6
	
	регистр состояния модема

	7
	
	регистр для временного хранения данных



Если работа с портом идет без использования прерываний, то их следует запретить путем записи 0 в регистр 1.

Ниже приведено описание битов регистров 3 и 5.

Регистр управления линией:

7 – DLAB (бит управления доступом к регистрам 0, 1): 1 – режим программирования делителя, 0 – чтение запись;

6 – бит формирования обрыва линии (вывод нулей);
5 – 0 – бит четности формируется автоматически, 1 – бит четности всегда равен инверсии 4-го бита;
4 – бит четности (1 – четность, 0 – нечетность);

3 – разрешение бита контроля четности (1 – да, 0 – нет);

2 – длина стоп-бита (0 – 1 бит, 1 – 2 бита);

1,0 – количество битов данных (00 – 5 бит, 01 – 6 бит, 10 – 7 бит, 11- 8 бит).

Регистр состояния линии:
7 – ошибка принятых данных;
6 – буфер передатчика пуст;

5 – регистр 0 передатчика пуст;

4 – индикатор обрыва линии;

3 – ошибка кадра (стоп-бит не найден);

2 – ошибка контрольного бита;

1 – ошибка переполнения, означает потерю символа (битов данных);

0 – регистр 0 приемника содержит принятые данные, сбрасывается при чтении этих данных.


Процедура инициализации порта COM2 может быть следующей:
     init PROC FAR

        mov dx, COM2_adr ;DX = базовый адрес порта COM2

        add dx, 3    ;DX = адресу регистра управления

        mov al, 80h  ;DLAB=1, настройка делителя

        out dx, al   ;Рабочая скорость - 4800 Кбод

        dec dx

        dec dx       ;DX = адресу старшего байта делителя скорости

        mov al, 0

        out dx, al

        dec dx       ;DX = адресу младшего байта делителя скорости

        mov al, 18h  ;Установка младшего байта делителя

        out dx, al

        add dx, 3    ;DX = адресу регистра управления

        mov al,00011011b ;DLAB=0,четность,1 стоп-бит,8 бит/слово;
        out dx, al

        dec dx

        dec dx       ;DX = адресу регистра разрешения прерываний

        mov al, 00h  ;Прерывания запрещены

        out dx, al

        ret
     init ENDP
Процедура передачи данных без контроля готовности приемника может быть следующей:

     send PROC FAR

        push ax      ;сохраняем АХ в стек

        mov dx, COM2_adr ;DX = базовому адресу COM2

        add dx, 5    ;DX = адресу регистра состояния

snd:    in al, dx    ;чтение байта состояния

        test al, 40h ;Бит 6=1(готов к передаче очередного байта ?)

        jz snd       ;Нет - ждем

        sub dx, 5    ;DX = адресу регистра передатчика

        pop ax       ;восстанавливаем AX 

        out dx, al   ;Передача

        ret

     send ENDP

Процедура приема данных может быть следующей:

     receive PROC FAR

        push ax          ;сохраняем АХ в стек

        mov dx, COM2_adr ;DX = базовому адресу COM2

        add dx, 5        ;DX = адресу регистра состояния

        in al, dx        ;чтение байта состояния

        test al, 01h     ;Бит 1 = "1" (принят байт) ?

        pop ax           ;восстанавливаем AX
        jz fin           ;Нет - выходим

        sub dx, 5        ;DX = адресу регистра приемника

        in al, dx        ;чтение принятого байта

fin:    RET

     receive ENDP
36. Пример 15. Модель телеуправления на диспетчерском щите

В качестве примера связи через последовательный интерфейс рассмотрим процедуру телеуправления тремя фидерами на диспетчерском щите:
     диспетчерский щит





ф3 подстанции
        ф1      ф2      ф3




На щите для каждого управляемого фидера имеется переключатель и индикатор состояния (если фидер выключен, то индикатор включен). При изменении коммутации фидера на диспетчерском щите формируется сигнал на подстанцию, содержащий номер фидера. На подстанции автоматически выполняется переключение фидера и формируется сигнал на диспетчерский щит о текущем положении выключателя (1 – включен, 2 – отключен), который воспринимается индикатором состояния и отображается на щите.

Для моделирования телеуправления использованы два компьютера, связанные через порт COM2. Один компьютер играет роль диспетчерского щита, второй – удаленную подстанцию. На первом компьютере выводится на экран начальное состояние трех фидеров. Переключение выполняется нажатием клавиш 1, 2, 3. Заданный номер фидера  посылается на вход второму компьютеру, который "выполняет переключение" и отсылает первому компьютеру новое состояние коммутации. При этом на обоих компьютерах отображается новое состояние фидера. На первом компьютере запущена процедура main, которая реагирует на нажатие клавиш 1,2,3. На втором компьютере запущена процедура fider, которая находится в постоянном ожидании приема номера фидера (1, 2, 3) из порта COM2. В обеих процедурах задействованы процедуры инициализации, приема и передачи сигналов через порт COM2. Выход из процедур осуществляется по клавише Esc. 
Программная реализация:
STK SEGMENT STACK
DB 64 DUP(?)

STK ENDS
DSEG SEGMENT
   COM2_adr DW 02F8h ;базовый адрес порта COM2
   fid DB 'Фидер 1 $','Фидер 2 $','Фидер 3 $' ;имена фидеров
   msg DB 'вкл',0Ah,0Dh,'$','отк',0Ah,0Dh,'$' ;индикатор состояния
   com DB 1,1,1 ;текущее состояние фидеров: 1-включен, 2-отключен
   sh1 DB 9 ;сдвиг по строкам массива fid
   sh2 DB 6 ;сдвиг по строкам массива msg
DSEG ENDS

CSEG SEGMENT

ASSUME CS:CSEG, DS:DSEG, SS:STK

init PROC FAR

...

init ENDP

send PROC FAR

...
send ENDP

receive PROC FAR

...
receive ENDP

out_fid PROC FAR ;вывод на экран фидера
XOR AH,AH   ;из массива fid
DEC AL      ;номер фидера задан в регистре AL
MUL sh1

LEA DX,fid  ;адрес строки в DS:DX
ADD DX,AX   ;позиционируемся на заданный фидер
MOV AH,09h  ;выводим на экран строку с '$' в конце
INT 21h

RET
out_fid ENDP

out_com PROC FAR ;вывод на экран состояния
XOR AH,AH   ;коммутации из массива com
DEC AL      ;коммутация задана в регистре AL
MUL sh2

LEA DX,msg
ADD DX,AX

MOV AH,09h

INT 21h

RET

out_com ENDP

fider PROC FAR     ;процедура на стороне подстанции
MOV AX,DSEG   ;(второй компьютер)
MOV DS,AX      ;инициализация сегмента данных

CALL init      ;инициализация СОМ порта

cont:
XOR AL,AL

CALL receive   ;чтение данных из порта
CMP AL,0    

JZ down        ;если ничего не принято то идем вниз

MOV BL,AL      ;если есть, то обработка сигала
CALL out_fid   ;вывод имени фидера
DEC BL      

MOV AL,com[BL] ;выборка текущего состояния коммутации
XOR AL,11b     ;инвертируем состояние ( AL
MOV com[BL],AL ;возвращаем новое состояние

CALL send      ;и отсылаем на первый компьютер (AL)
CALL out_com ;выводим на экран новое соcтояние коммутации
down:
MOV AH,06h     ;ввод символа без ожидания

MOV DL,0FFh

INT 21h

JZ  cont  ;если символ не введен то продолжим опрос порта

CMP AL,1Bh ;если Esc то выходим

JNE cont   ;если другой символ то продолжим опрос порта
MOV AX,4C00h ;завершение процедуры
INT 21h

fider ENDP

main PROC FAR   ;процедура на стороне диспетчерского щита
MOV AX,DSEG ;(первый компьютер)
MOV DS,AX   ;инициализация сегмента данных

CALL init   ;инициализация СОМ порта

MOV CX,3    ;начальный вывод состояния фидеров

nxt:

MOV AX,CX

CALL out_fid ;вывод фидера
MOV SI,CX

DEC SI

MOV AL,com[SI] ;выборка коммутации
CALL out_com ;вывод коммутации фидера на экран
LOOP nxt    

rpt:

MOV AH,08h  ;чтение символа с ожиданием
INT 21h

CMP AL,1Bh  ;если Esc то выход

JE fin1

SUB AL,30h  ;переводим коды символов 1,2,3 в цифры

JZ rpt      ;если 0 то вводим новый символ

JS rpt      ;если < 0 то вводим новый символ

CMP AL,3

JG rpt      ;если > 3 то вводим новый символ

CALL send    ;отправка фидера (AL) на компьютер 2
CALL out_fid  ;вывод строки фидера (AL)
rcv:

XOR AX,AX    ;ожидание подтверждения изменения коммутации

CALL receive ;чтение порта
CMP AL,0    

JZ rcv      ;если ничего не принято то ждем
CALL out_com ;выводим измененную коммутацию

JMP rpt
fin1:
MOV AX,4C00h ;завершение процедуры
INT 21h

main ENDP

CSEG ENDS   

END main – для компьютера 1, fider – для компьютера 2
Комментарий:

1. Номер фидера и состояние коммутации находятся в регистре AL.


2. Ожидание приема проверяется по условию AL = 0, если AL ≠ 0, то данные приняты.


3. Процедура send не затирает регистр AX, а процедуры out_fid, out_com затирают, поэтому в реализации сначала вызывается процедура send.

4. В процедуре main использована функция чтения символа с ожиданием (AH = 08h), а в процедуре fider без ожидания (AH = 06h, DL = FFh), так как в ней должен постоянно опрашиваться порт COM2.

5. В последней строке реализации, после команды END для компьютера 1 указывается точка входа main, а для компьютера 2 – fider. 


6. Отображение работы программы на обоих компьютерах может быть следующим:


Компьютер 1 (диспетчерский щит)

Компьютер 2 (подстанция)

загрузка одного слова из поля NAMЕ1 в аккумулятор AX. Команда LODSW выполняется без префикса повторения





в поле NAME 10 байт заполняются символом " "





здесь выполняется сравнение двух строк NAME1 и NAME2 длинной в 10 байт и после выхода будет выставлен соответствующий флаг. Ветвление по результату реализуется операцией JNE





сравнение байт аккумулятора и памяти; Выполняется поиск в строке NAME1 длинной в 10 байт символа М; после выхода, если символ найден его позиция зафиксирована в DI





Здесь в строке STRING длинной в 15 байт выполняется поиск символа &. Если он найден, с помощью абсолютной адресации по регистру DI, он заменяется пробелом.





Выполняется заполнение строки POLE, в результате чего она будет иметь вид:


POLE='***---***---***'


Реализация требует расположения  строк FRAG и POLE подряд.

